IDENTIFICATION

PRODUCT CODE:	DEC-08-EUFB-D
PRODUCT NAME:	TC01-TU55 DECTAPE FORMATTER
DATE CREATED:	APRIL 9, 1970
MAINTENANCE CONTRACT	SOFTWARE INFORMATION SERVICE (CDB)

1. ABSTRACT

This program (TOG-8) records the required timing and mark tracks on a DECtape mounted on the TC01-TU55 DECtape unit.

The program interacts with you via the ASR-33 Teletype[1] to obtain the necessary data for each set of DECtapes to be formatted. As soon as one set of tapes is formatted, the program is ready to format another set.

Two full passes are required to completely format each DECtape, and up to eight DECtapes may be formatted at a time (assuming that the user has eight tape transports). Upon completion of a cycle, new tapes may be mounted and formatted as the last, with a minimum of operator-program communication. One tape, excluding tape setup time, requires two minutes from start to finish.

2. PRELIMINARY REQUIREMENTS

2.1 Equipment

PDP-8(R), ASR-33 Teletype, TC01-TU55 DECtape Control.

2.2 Storage

TOG-8 requires 2500_8 registers.

3. LOADING PROCEDURE

Load the program into core using the standard Binary Loader.

4. USING THE PROGRAM

4.1 Starting Procedure

a. Key 1000 into the SWITCH REGISTER. Depress LOAD ADDRESS and depress START. "DTA?" is printed on the ASR 33.

Mount the DECtapes to be marked onto the tape transports, with just enough turns of tape on the right hand reel of each transport to provide a grip. Make sure that no two tape units are set to the same unit number. Set the RDMK-WRTM-NORMAL switch located on the TC01 maintenance control panel to the WRIM position; for each transport to be used, set the WRITE ENABLED-WRITE LOCK switch to WRITE ENABLED, and the REMOTE-OFF-LOCAL switch to REMOTE.

[1]Teletype is a registered trademark of the Teletype corporation.

(R)PDP is a registered Trademark of the Digital Equipment Corporation.

4.2 Operating Procedures

The program and operator now converse. The printout "DTA?" is asking which DECtape units will be used. The operator types a unit number or series of unit numbers, corresponding to the DECtape units upon which he has mounted tapes. For instance, if the operator has mounted tapes on units 2, 5, 7, and 8, he would type 2 5 7 8↵ (where ↵ signifies carriage return). Spaces are ignored, so it makes no difference if the operator types spaces between the unit numbers. Only one specification of a unit is significant, i.e., typing 2 2 5 7 7 5 8 2 8↵ has the same effect as typing 2 5 7 8↵ .

Once the operator has specified the units he wishes to use, the program types "DIRECT?" The operator responds by typing MARK↵ or MARK XXXX↵ . If he types MARK↵ , the program assumes 201_8 words, 2702_8 blocks (standard PDP-8 format). Otherwise, XXXX is accepted as a decimal number of words per block, and must be divisible by 3. Note that typing MARK 384↵ will cause the program to generate a standard PDP-10 format DECtape (1102_8 blocks of 600_8 words, which is equivalent to 1102_8 blocks of 200_8 words, were each word is 36 bits rather than 12 bits).

The program now types "XXXX WORDS, YYYY BLOCKS OK? (YES OR NO)." This serves as a final check for block count. XXXX and YYYY are octal values representing the final outcome of a formula solved by the program, determining the number of blocks that may be written on a DECtape knowing the number of words. If a NO↵ answer is given, the program reverts to "DIRECT?" Otherwise (if YES↵), the tape on the first unit specified begins to move.

Once all of the tapes specified have been marked, the printout "SET SWITCH TO NORMAL" appears. Then the operator returns the "RDMK-WRTM-NORMAL" switch to NORMAL, and strikes the RETURN key on the ASR-33, starting the second pass. Note that during the second pass with multiple DECtape units, as soon as one tape stops and the next tape starts, the first tape is completed and may be replaced with a fresh tape in preparation for recycling.

The program continues by itself until completed, at which time the "DIRECT?" printout occurs. Typing "SAME↵" repeats the entire process with the original constants. The new DECtapes must be mounted and ready to write timing and mark tracks before "SAME↵" is

typed. Also, in response to "DIRECT?", typing "RDR↵" causes the printout of the unit numbers of the DECtapes and the last twelve block numbers; "RDF↵" causes the printout of the unit numbers and the first twelve block numbers; and "RESTART↵" returns the program to "DTA?" Unit numbers are printed as "N000", where N is the unit number (0 means DECtape unit 8). If the ION lamp on the PDP-8 console is lit, typing "CONTROL C" causes the program to restart at "DTA?"

Following are several examples of successful operation. The underlined statements are printed by the program. ALL operator responses should be followed by a carriage return.

a. Create a standard tape on unit 4.

```
DTA? 4
DIRECT? MARK
Ø2Ø1 WORDS, 27Ø2 BLOCKS.OK? YES OR NO
YES
SET SWITCH TO NORMAL
DIRECT?
```

b. Create 16 standard PDP-10 format tapes - eight at a time, on units 1 - 8.

```
DTA? 12345678
DIRECT? MARK 384
Ø6ØØ WORDS, 11Ø2 BLOCKS OK? YES OR NO
YES
SET SWITCH TO NORMAL (USER TYPES ↵ )
DIRECT? SAME
SET SWITCH TO NORMAL (USER TYPES ↵ )
DIRECT?
```

4.3 Errors

4.3.1 Errors Typed to "DTA?" and "DIRECT?" - Revert back to "DTA?" or "DIRECT?"

4.3.2 Error Messages for Response to MARK XXXX -

NOT DECIMAL	A character in XXXX is not 0-9.
NOT DIVISIBLE BY 3	XXXX cannot be divided evenly by 3.
TOO MANY WORDS	The number of words plus 15 exceeds 7777_8.
TOO MANY BLOCKS	The number of blocks generated by XXXX exceeds 7777_8.

4.3.3 Error Messages for Response to YES (After message - revert back to "DTA?")

SETUP?	Indicates an error in the DECtape setup -
	Unit in WRITE LOCK
	Nonselectable unit
	Switch not in WRTM position

4.3.4 Error Messages for Marking and Verifying a Tape

```
XXXX SHOULD BE YYYY BLK ERROR PHASE X
XXXX SHOULD BE YYYY DATA ERROR PHASE X
END TAPE ERROR PHASE X
MARK TRACK ERROR PHASE X
PARITY ERROR PHASE X
SELECT ERROR PHASE X
TIMING ERROR PHASE X
LAST INT NOT END ZONE (see 4.4)
```

4.4 Recovery

Although error should cause doubt concerning the entire process, restarts may be made by phases (except when in phase 0). Restart the phase by typing "RETRY)." Type "RESTART" to return to "DTA?"

PHASE 0:	MARK TRACK WRITE
PHASE 1:	WRITING LAST REVERSE BLOCK NUMBER FORWARD
PHASE 2:	WRITING BLOCK NUMBERS AND DATA IN REVERSE
PHASE 3:	READING AND CHECKING BLOCK NUMBERS AND DATA

An error that should be considered catastrophic is LAST INT NOT END ZONE. This indicates that between the last (or first) block number and the end zone, something caused an interrupt (DTF).

The entire program may be restarted at 1000_8 any time.

5. DETAILS OF OPERATION AND STORAGE

The program writes timing and mark tracks on a DECtape, then inserts block numbers and parity correct information, checking the results of all operations.

The number of block frames to be written is a function of the number of words per block. The formula

$$\text{blocks per tape} = \frac{212080}{NW+15} + 2$$

where NW equals the number of words to be written, is used by the program to compute the number of blocks, but is adjusted by the program to provide the standard PDP-8 format of 129 (12-bit) words, 1744 blocks, and standard PDP10 format of 128 (36-bit) words, 578 blocks.

Two full passes are required to mark and verify a tape.

Pass 1	Marks the tape forward, inserts block numbers and parity correct data in reverse.
Pass 2	Reads and checks block numbers and data forward and reverse.

During the forward direction of the first pass, the TC01 is switched into WRITE TIMING AND MARK TRACKS, CONTINUOUS MODE, FORWARD. The program manipulates data to be written by monitoring the word count register and the DTF, (DECtape flag). Initially, ten feet of end-zone code is written, and abutting the end zone are about two standard block lengths of interblock sync. To the TC01, this interblock sync acts as no operation, but guarantees that at turn-around time, block 0 is read first (or 2701 if turning out of the forward end zone). Now the remainder of the tape is written creating block frames. The number of such frames is determined by the above formula. Upon completion of the block framing, another extended interblock sync zone is written as well as ten feet of end zone.

Pass 1 forward is now complete (timing and mark tracks are written). The tape is ordered to MOVE in reverse for three seconds, thus moving it out of the end zone and onto the marked section. The tape is once again moved forward, and the last REVERSE BLOCK NUMBER is written until the forward end zone is sensed. Now the tape is turned out of the end zone in SEARCH, and the program waits for a block interrupt (first reverse block number). When the DTF rises, the TC01 is switched into WRITE ALL, CONTINUOUS, REVERSE; thus the system is synchronized and all block numbers and data are written until the forward end zone is sensed. This completes the marking and blocking of the tape. Pass 2 in CONTINUOUS MODE checks the data and block numbers to be certain they are correct. When multiple DECtape units are specified, Pass 1 forward is completed for each tape before Pass 1 reverse is begun.

5.1 Theory

The program flow is based on the following detailed description of the bit structure on the mark track, followed by a description of block and data writing.

a. Install the tape with enough turns to create a pull. The reverse end zone requires a sequence of three data words for its pattern.

4044
0440
4404

In the mark track the words appear as 101101101101 (5555_8). The reverse end zone should cover 10 feet of tape. Write the above three words 4096_{10} times.

b. 100 interblock sync (see c.).

c. Interblock sync. Three words of interblock sync should immediately follow each block.

0404
0404
0404

In the mark track the words appear as 010101010101 (2525_8).

d. The forward block mark and reverse guard require three words.

0404
4004
4040

which appear on the mark track as 010110011010 (2632_8).

block number frames

e. The lock mark, reverse checksum, reverse final, and reverse prefinal consist of six PDP-8 memory words.

```
0040
0000
4000
0040
0000
4000
```

These words appear on the mark track as



f. Mark track code for data is generated by

```
4440
0044
4000
```

These three words appear as 111000111000(7070_8) and are repeated 41_{10} times for a 129 word block.

g. The prefinal, final, checksum, and reverse lock consist of six PDP-8 words.

```
4440
4444
4044
4440
4444
4044
```

These words appear in the mark track as



h. The guard and reverse block number consist of three words

4040
0440
0404

which appear as 101001100101 (5145_8)
reverse block number

i. Generate 2702_8 block patterns. Repeat c through h, 2702_8 times.

j. 100 more interblock syncs (see c.).

k. The end zone pattern consist of three words

0400
4004
0040

which appear as 010010010010 (2222_8). Repeat these three words 4096 times.

5.2 Format Block Numbers and Data

Once the mark track has been prepared for each tape specified block numbers and data must be written.

a. With tape in the forward end zone reverse tape for a few seconds to move the head onto the mark track.

b. Go forward in SEARCH waiting for a DTF.

c. When the DTF rises, switch to READ DATA with the word count register equal to one less than the total number of words in the block. When the word count goes to zero, write 12_{10} words (see figure 1) including block number (last reverse) in WRITE ALL. When the word count goes to zero, go back to step c.

d. Continue this process until the forward end zone is sensed.

e. Reverse tape now in SEARCH and wait for the DTF. When the DTF rises, switch to WRITE ALL CONTINUOUS. The system remains in WRITE ALL.

f. Write three words.

```
V1      0000
V2      0000
V3      0077    77 = forward checksum
```

When the word count register goes to zero, reload it with one less than the total words to be written and set the current address counter to the address of three words of all sevens. Monitor the word count until it goes to zero. Keep resetting the current address counter to the address of three words of all sevens.

When the word count goes to zero, write the twelve words (figure 2) shown below and return to write the three words again. Continue this process until the reverse end zone is reached.

```
Core                    Direction of write forward
          0000
          0000
          0000          Assuming that the last block number is
          0000          1763, V1 and V2 constitute the last re-
V1        0041          verse block number.  When read in reverse,
V2        0677          4106 becomes 1763.  Remember that a 1
          7777          written forward becomes a 0 when read
          7700          in reverse, and a 0 becomes a 1.  (The
          0000          trailing 7's are necessary so that the
          0000          36 bit block number used on the PDP-10,
          0000          will be read in reverse with leading 0's).
          0000
                        Direction of read reverse
```

Figure 1 Last Reverse Block Number as Written Forward

```
Core
          CC00          CC=reverse checksum should be 00
          0000
          0000
          0000          V1 and V2 comprise the forward block
V1        0041          number.  V3 and V4 comprise the reverse
V2        0677          block number.  Therefore, when V1 and
          7777          V2 are read forward, the resulting block
          7700          number is 1763.  Remember that a 1 writ-
          0000          ten in reverse is read forward as a 0,
V3        0017          and a 0 as a 1.  V3 and V4 are read in
V4        6200          reverse as written, 1762.
          0000
```



Figure 2 Block Numbers Written in Reverse

```
                        /COPYRIGHT 1970 DIGITAL EQUIPMENT CORP.
                        /MAYNARD, MASS.
                        /REVISED MARCH 1970

                        /   TOG-8       TO MARK AND CHECK PDP-8 DECTAPE
                        /THIS PROGRAM WRITES TIMING AND MARK TRACKS ON
                        /DECTAPE MOUNTED ON THE TC01-TU55 TAPE CONTROL UNIT.



                  0010          X1=10
                  0011          X2=11

                        /SYMBOL TABLE AUGMENTATION

                  6761          DTRA=6761
                  6762          DTCA=6762
                  6764          DTXA=6764
                  6771          DTSF=6771
                  6772          DTRB=6772
                  6774          DTLB=6774
                  6766          DTCX=6766

                        /SET 0 FOR THE LOGIN FEATURE

                  0000          *0
0000  0000                      0
0001  5402                      JMP I .+1
0002  0543                      CONC    /CONTROL "C" AND LOGIN

                        /WORKING LOCATIONS

                  0020          *20

0020  0000     W1,     0000
0021  0000     W2,     0000
0022  0000     W3,     0000
0023  0000     W4,     0000
0024  0000     W5,     0000
0025  0000     W6,     0000
0026  0000     BLOCKS, 0000
0027  0000     BLOCKA, 0000
0030  0000     DTA,    0000
0031  0000     ERX,    0000
0032  0000     PHASE,  0000
0033  0000     TOTAL,  0000
0034  0000     VAR1,   0000
0035  0000     VAR2,   0000
```

/CONSTANTS

```
0036  0001    C1,      0001
0037  0002    C2,      0002
0040  0003    C3,      0003
0041  0004    C4,      0004
0042  0017    C0017,   0017
0043  0070    C0070,   0070
0044  0077    C0077,   0077
0045  0007    C0007,   0007
0046  0030    C0030,   0030
0047  0400    C0400,   0400
0050  0700    C0700,   0700
0051  0203    C203,    0203
0052  0201    C201,    0201
0053  0210    C210,    0210
0054  0260    C260,    0260
0055  0261    C261,    0261
0056  0267    C267,    0267
0057  0270    C270,    0270
0060  0271    C271,    0271
0061  0277    C277,    0277
0062  1000    C1000,   1000
0063  1620    C1620,   1620
0064  7000    C7000,   7000
0065  7700    C7700,   7700
0066  7714    C7714,   7714
0067  7761    C7761,   7761
0070  7772    C7772,   7772
0071  7775    C7775,   7775
0072  0215    CRCOD,   0215
0073  0313    LETK,    0313
0074  0212    LFCOD,   0212
0075  7776    M2,      -2
0076  7775    M3,      -3
0077  7774    M4,      -4
0100  7772    M6,      -6
0101  7771    M7,      -7
0102  7764    M14,     -14
0103  7634    M144,    -144
0104  7500    M300,    -300
0105  0240    SPCOD,   0240
```

/INTERPAGE LINKS

```
0106  0020    ADW2,   W2-1
0107  0021    ADW3,   W3-1
0110  2452    BADD,   BUFFER-1
0111  2453    BFR,    BUFFER
0112  7755    CA,     7755
0113  0310    COMPAR, COMPRE
0114  0000    FCON,   0000
0115  1055    IT,     INIT1
0116  0526    FORMA,  FORM-1
0117  0527    FORMB,  FORM
0120  0400    QU1,    Q1
0121  0410    QU2,    Q2
0122  0422    QU3,    Q3
0123  0434    QU4,    Q4
0124  0454    MESS,   MES
0125  0633    MWAIT,  MINI
0126  1000    STX,    START
0127  0613    TURN,   TRN
0130  0334    TYOCT,  TYCT
0131  0200    TYPE,   MESAGE
0132  0256    TYPIN,  TYPN
0133  0600    WAIT,   STALL
0134  7754    WC,     7754
```

```
                        /TYPE THE CHARACTER IN THE AC ON THE KEYBOARD PRINTER

0135  0000    RSEND,  0000
0136  6046            TLS             /LOAD AND PRINT, CLEAR FLAG
0137  6041            TSF             /WAIT FOR CONFIRMATION
0140  5137            JMP    .-1      /ENDLESSLY
0141  6042            TCF             /CLEAR THE FLAG ANYWAY
0142  5535            JMP I RSEND

                        /PRINT A "?" ON THE KEYBOARD TYPER

0143  0144    QU,     .+1
0144  6002            IOF             /KILL LOG AND CONTROL C FCTN
0145  7300            CLA   CLL       /C(AC)+C(L)=0
0146  1061            TAD   C277      /"?"
0147  4135            JMS   RSEND     /TYPE THE CHARACTER
0150  5551            JMP I .+1       /RESTART
0151  1061            INIT

                        /DECTAPE CONTROL WORDS

0152  0030    DT0030, 0030
0153  0060    DT0060, 0060
0154  0070    DT0070, 0070
0155  0100    DT0100, 0100
0156  0130    DT0130, 0130
0157  0140    DT0140, 0140
0160  0200    DT0200, 0200
0161  0210    DT0210, 0210
0162  0360    DT0360, 0360
0163  0510    DT0510, 0510
0164  0600    DT0600, 0600
0165  0610    DT0610, 0610

                        /SOME SPECIAL LINKS

0166  2475    ADBA,   2475
0167  2476    ADWA,   2476
0170  2477    ADWAB,  2477

                        /CONSTANTS FOR FORMULA TRANSLATION SECTION

0171  0172    BINCON, .+1
0172  0001            0001
0173  0012            0012
0174  0144            0144
0175  1750            1750
```

```
                0200            *200    /PAGE 1
                                /TYPE CANNED MESSAGES.....
                                /THANKS TO DIGITAL 8-18-U

0200  0000      MESAGE, 0
0201  6002              IOF             /KILL LOG AND CONTROL FUNCTION
0202  7240              CLA   CMA       /SET C(AC)=-1
0203  1200              TAD   MESAGE    /ADD LOCATION
0204  3010              DCA   10        /AUTO INDEX REGISTER
0205  1410              TAD I 10        /FETCH FIRST WORD
0206  3217              DCA   MSRGHT    /SAVE IT
0207  1217              TAD   MSRGHT
0210  7012              RTR
0211  7012              RTR             /ROTATE 6 BITS TO THE RIGHT
0212  7012              RTR
0213  4220              JMS   TYPECH    /TYPE IT
0214  1217              TAD   MSRGHT    /GET DATA AGAIN
0215  4220              JMS   TYPECH    /TYPE RIGHT HALF
0216  5205              JMP   MESAGE+5  /CONTINUE
0217  0000  MSRGHT, 0                  /TEMPORARY STORAGE
0220  0000  TYPECH, 0                  /TYPE CHARACTER IN C(AC)6-11
0221  0044              AND   C0077
0222  7450              SNA             /IS IT END OF MESSAGE?
0223  5410              JMP I 10        /YES: EXIT
0224  1251              TAD   M40       /SUBTRACT 40
0225  7500              SMA             /<40?
0226  5231              JMP   .+3       /NO
0227  1252              TAD   C340      /YES: ADD 300
0230  5244              JMP   MTP       /TO CODES <40
0231  1076              TAD   M3        /SUBTRACT 3
0232  7441              SZA             /IS IT ZERO?
0233  5236              JMP   .+3       /NO
0234  1253              TAD   C212      /YES: CODE 43 IS
0235  5244              JMP   MTP       /LINE-FEED (212)
0236  1075              TAD   M2        /SUBTRACT 2
0237  7441              SZA             /IS IT ZERO?
0240  5243              JMP   .+3       /NO
0241  1254              TAD   C215      /YES: CODE 45 IS
0242  5244              JMP   MTP       /CARRIAGE RETURN (215)
0243  1255              TAD   C245      /ADD 200 TO OTHERS >40
0244  6046  MTP,        TLS             /TRANSMIT CHARACTER
0245  6041              TSF             /WAIT FOR THE FLAG
0246  5245              JMP   .-1       /NOT SET YET
0247  7200              CLA             /SET: CLEAR C(AC)
0250  5620              JMP I TYPECH    /RETURN

                        /CONSTANTS

0251  7740  M40,    -40
0252  0340  C340,   340
0253  0212  C212,   212
0254  0215  C215,   215
0255  0245  C245,   245
```

```
                        /ROUTINE WAITS UNTILL A COMPLETE MESSAGE HAS BEEN ENTERED
                        /SIGNIFIED BY A CR.

0256  0000    TYPN,   0
0257  6002            IOF             /KILL THE LOG AND CONTROL C FUNCTION
0260  6032            KCC             /CLEAR AC, KEYBOARD FLAG
0261  1110            TAD   BADD      /GET BUFFER ADDRESS
0262  3020            DCA   W1        /STORE FOR THE CHARACTER STRING

                        /READ AND RESPOND WITH THE CHARACTER

0263  2020    NTYRTN, ISZ   W1        /NORMAL RETURN, INCREMENT BUFFER
0264  6031            KSF             /WAIT FOR KEYBOARD
0265  5264            JMP   .-1       /FLAG TO RAISE
0266  6036            KRB             /GOT FLAG, RESET IT, GET CHARACTER
0267  4135            JMS   RSEND     /SEND CHARACTER BACK
0270  3420            DCA I W1        /LOAD CHARACTER INTO BUFFER AREA

                        /IF CHARACTER IS A SPACE, IGNORE IT

0271  1420            TAD I W1        /CHARACTER INTO THE AC
0272  7041            CIA             /SUBTRACT FROM SPACE CODE (240)
0273  1105            TAD   SPCOD     /COMPLETE COMPARISON
0274  7650            SNA   CLA       /WAS IT A SPACE?
0275  5264            JMP   NTYRTN+1  /YES: DO NOT INCREMENT BUFFER

                        /IF CHARACTER IS A CR, EXIT FROM ROUTINE

0276  1420            TAD I W1        /CHARACTER TO AC
0277  7041            CIA             /SET AC TO SUBTRACT CR (215)
0300  1072            TAD   CRCOD     /COMPLETE COMPARISON
0301  7640            SZA   CLA       /WAS IT CR?
0302  5263            JMP   NTYRTN    /NO: INCREMENT BUFFER + WAIT

                        /CARRIAGE RETURN FOUND, EXIT FROM ROUTINE

0303  1074            TAD   LFCOD     /GIVE KEYBOARD LINE FEED
0304  4135            JMS   RSEND     /EXECUTE LINE FEED
0305  7300            CLA   CLL       /EXIT WITH C(ACC) = AND C(L)=0
0306  6001            ION             /RESET LOG AND CONTROL C FUNCTION
0307  5656            JMP I TYPN      /RETURN TO CALL
```

```
                        /COMPARE A STRING OF CHARACTERS IN "BUFFER"
                        /TO A CHARACTER STRING AFTER A JMS IN ASCII

0310  0000    COMPRE, 0
0311  7240            CLA   CMA       /C(AC)=7777
0312  1310            TAD   COMPRE    /SUBTRACT 1 FOR INDEX REG 1
0313  3010            DCA   10        /AUTO INDEX 1 SET TO CHA STRING
0314  1110            TAD   BADD      /AUTO INDEX 2 SET TO BUFFER-1
0315  3011            DCA   11        /LOAD X2

                        /COMPARE CHARACTERS TILL ONE DOESN'T COMPARE OR TILL
                        /A 0 IS FOUND IN X1. IF OK, RETURN TO TWO PLUS THE
                        /ZERO, IF BAD ONE PLUS

0316  1410            TAD I X1        /CHARACTER FROM PROGRAM
0317  7041            CIA             /TO SUBTRACT FROM
0320  1411            TAD I X2        /CHARACTER IN BUFFER
0321  7640            SZA   CLA       /COMPARE?
0322  5330            JMP   CERR      /NO:RESYNC FOR NON COMPARE EXIT
0323  1410            TAD I X1        /YES: CHECK FOR GOOD EXIT
0324  7440            SZA             /IF 0, EXIT GOOD
0325  5317            JMP   .-6       /NO: TEST NEXT CHAACTER
0326  2010            ISZ   X1        /+1 TO X1(TOTAL 2 FROM THE 0)
0327  5410            JMP I X1        /+1 TO X1, EXIT

                        /ERROR FOUND, RESYNC AND EXIT NO COMPARE

0330  1410    CERR,   TAD I X1        /CHARACTER FROM PROGRAM
0331  7640            SZA   CLA       /IS THIS EXIT KEY? (0000)
0332  5330            JMP   .-2       /NO: GET NEXT
0333  5410            JMP I X1        /YES: EXIT, NOT COMPARE
```

```
                        /TYPE ONE FOUR CHARACTER OCTAL WORD GIVEN TO THE
                        /ROUTINE VIA C(ACC), C(ACC)=0 ON EXIT

0334  0000    TYCT,   0
0335  3374            DCA   TW1         /STORE WORD GIVEN
0336  1374            TAD   TW1         /TO C(ACC) AGAIN
0337  7012            RTR
0340  7012            RTR               /6 BITS GIGHT
0341  7012            RTR
0342  3371            DCA   TYCT1+2     /SAVE ROTATED VALUE, 1ST TWO
0343  1371            TAD   TYCT1+2     /TO C(ACC) AGAIN
0344  0045            AND   C0007       /ISOLATE SECOND CHARACTER
0345  1375            TAD   C6060       /CONVERT TO ASCII
0346  3370            DCA   TYCT1+1     /STORE AS FIRST PARTIAL 2
0347  1371            TAD   TYCT1+2     /ROTATED VALUE STORED ABOVE
0350  7006            RTL
0351  7004            RAL               /3 BITS LEFT
0352  0050            AND   C0700       /ISOLATE FIRST CHARACTER
0353  1370            TAD   TYCT1+1     /CONVERT 1ST TO ASCII
0354  3370            DCA   TYCT1+1     /1ST AND 2ND CHARACTERS READY
0355  1374            TAD   TW1         /ORIGIONAL WORD
0356  0045            AND   C0007       /ISOLATE 4TH CHARACTER
0357  1375            TAD   C6060       /CONVERT 4 TH TO ASCII
0360  3371            DCA   TYCT1+2     /STORE 4TH FOR A MOMENT
0361  1374            TAD   TW1         /ORIGIONAL WORD
0362  7006            RTL
0363  7004            RAL               /POSITION IT 3RD CHARACTER
0364  0050            AND   C0700       /ISOLATE 3RD CHARACTER
0365  1371            TAD   TYCT1+2     /CONVERT TO ASCII
0366  3371            DCA   TYCT1+2     /CONVERSION COMPLETE
0367  4531    TYCT1,  JMS I TYPE        /TYPE THE FOUR CHARACTERS
0370  0001            0                 /FIRST 2
0371  0007            0                 /SECOND 2
0372  0000            0                 /KILL KEY
0373  5734            JMP I TYCT        /EXIT FROM ROUTINE

                        /SOME CONSTANTS FOR THE ROUTINE

0374  0000    TW1,    0000
0375  6060    C6060,  6060
```

```
                0400            *400
                                /VARIOUS ERROR MESSAGES
                                /"NOT DECIMAL"

0400  4531       Q1,     JMS I TYPE
0401  1617               1617  /NO
0402  2440               2440  /T
0403  0405               0405  /DE
0404  0311               0311  /CI
0405  1501               1501  /MA
0406  1403               1400  /L
0407  5247               JMP   QUX

                                /"TO MANY WORDS"

0410  4531       Q2,     JMS I TYPE
0411  2417               2417  /TO
0412  1740               1740    /O
0413  1501               1501    /MA
0414  1631               1631    /NY
0415  4027               4027    / W
0416  1722               1722    /OR
0417  0423               0423    /DS
0420  0000               0000    /00
0421  5247               JMP   QUX

                                /"TO MANY BLOCKS"

0422  4531       Q3,     JMS I TYPE
0423  2417               2417  /TO
0424  1740               1740    /O
0425  1501               1501    /MA
0426  1631               1631    /NY
0427  4002               4002    / B
0430  1417               1417    /LO
0431  0313               0313    /CK
0432  2300               2300    /S0
0433  5247               JMP   QUX

                                /"NOT DIVISIBLE BY 3"
0434  4531       Q4,     JMS I TYPE
0435  1617               1617  /NO
0436  2440               2440  /T
0437  0411               0411  /DI
0440  2611               2611  /VI
0441  2311               2311  /SI
0442  0214               0214  /BL
0443  0540               0540  /E
0444  0231               0231  /BY
0445  4063               4063  / 3
0446  0000               0000  /00
0447  4531       QUX,    JMS I TYPE
0450  4345               4345  /CR*LF
0451  0000               0000  /END
0452  5653               JMP I .+1
```

0453 1061 INIT

```
                /THE CODING BELOW CREATES THE BLOCK NUMBER
                /CONVERSION PRIOR TO THE TAPE WRITE.

0454  0000      MES,    0
0455  3023              DCA   W4          /BLOCK NUMBER GIVEN IN AC
0456  1023              TAD   W4          /RESTORE TO AC AGAIN
0457  7040              CMA               /COMPLEMENTED
0460  7006              RTL
0461  7004              RAL               /LEFT 3
0462  3024              DCA   W5          /TEMP SAVE
0463  1024              TAD   W5          /TO AC AGAIN
0464  0064              AND   C7000       /ISOLATE HIGH CHA
0465  3334              DCA   V2          /FORWARD BLOCK NUMBER
0466  1024              TAD   W5          /SHIFTED VALUE
0467  0043              AND   C0070       /ISOLATE 6,7,8
0470  3333              DCA   V1          /FORWARD BLOCK NUMBER
0471  1023              TAD   W4          /ORIGIONAL SET
0472  7040              CMA               /UPSIDE DOWN
0473  7012              RTR
0474  7010              RAR               /RIGHT 3
0475  3024              DCA   W5          /TEMP SAVE
0476  1024              TAD   W5          /TO AC AGAIN
0477  0050              AND   C0700       /ISOLATE 3,4,5
0500  1334              TAD   V2          /COMBINE FORWARD BLOCK NUMBER
0501  1044              TAD     C0077
0502  3334              DCA   V2          /1/2 COMPLETE
0503  1024              TAD   W5          /SHIFTED VALUE
0504  0045              AND   C0007       /ISOLATE 9, 10,11
0505  1333              TAD   V1          /COMBINE WITH 8N
0506  3333              DCA   V1          /FORWARD BLOCK NUMBER COMPLETE

                /CONVERT REVERSE BLOCK NUMBER

0507  7040              CMA               /-1 TO GIVEN BLOCK #
0510  1023              TAD   W4          /ORIGIONAL BLOCK #
0511  3024              DCA   W5          /TEMP SAVE
0512  1024              TAD   W5          /TO AC AGAIN
0513  7012              RTR
0514  7012              RTR               /6 RIGHT
0515  7012              RTR
0516  0044              AND   C0077       /ISOLATE LOW
0517  3340              DCA   V3          /HIGH REVERSE
0520  1024              TAD   W5          /COMPLEMENT ORIGIONAL -1
0521  7006              RTL
0522  7006              RTL               /6 LEFT
0523  7006              RTL
0524  0065              AND   C7700       /ISOLATE HIGH
0525  3341              DCA   V4          /REVERSE COMPLETED
0526  5654              JMP I MES
```

/FORM USED TO WRITE 12 DATA WORDS FOR BLOCK NUMBERING

```
0527  0000  FORM,   0000
0530  0000          0000
0531  0000          0000
0532  0000          0000
0533  0000  V1,     0000
0534  0000  V2,     0000
0535  7777          7777
0536  7700          7700
0537  0000          0000
0540  0000  V3,     0000
0541  0000  V4,     0000
0542  0000          0000
```

```
                        //THIS ROUTINE ALLOWS KEYBOARD INTERRUPTION
                        /FOR LOGGING ON THE KEYBOARD, OR FOR A MAJOR
                        /CLEAR IN THE PROGRAM. BY HITTING "CONTROL C"
                        /A SYSTEM RESTART WILL OCCUR.

0543  6041    CONC,   TSF             /IS THE PRINTER FLAG ON?
0544  5346            JMP   .+2       /NO, CHECK READER
0545  6042            TCF             /YES; RESET IT
0546  6031            KSF             /IS THE READER FLAG ON?
0547  5364            JMP     RSYC-2  /NO: RETURN TO SEQUENCE

                        /OK. CHECK FOR EITHER LOG OR CONTROL C.

0550  3254            DCA   MES       /SAVE C(AC)
0551  7004            RAL             /SAVE THE LINK
0552  3374            DCA   RSYC+6    /FOR LOGGING
0553  6036            KRB             /GET CHARACTER FROM KEYBOARD
0554  6046            TLS             /RETURN CHARACTER
0555  7041            CIA             /TO SEE IF
0556  1251            TAD   C203      /"CONTROL C"
0557  7650            SNA   CLA       /IS IT?
0560  5366            JMP   RSYC      /YES; RESYNC THE PROGRAM
0561  1374            TAD   RSYC+6    /RESTORE THE LINK
0562  7010            RAR             /FOR EXIT.
0563  1254            TAD   MES       /THE AC TOO
0564  6001            ION             /INTERRUPT ON
0565  5407            JMP I Ø         /RETURN

                        /RESYNC THE SYSTEM TO START

0566  6041    RSYC,   TSF             /WAIT FOR FLAG
0567  5366            JMP   .-1       /ON LAST SENDOFF
0570  4531            JMS I TYPE
0571  2205            2205  /RE
0572  2331            2331  /SY
0573  1603            1603  /NC
0574  0000            0000  /END
0575  1030            TAD   DTA       /TO KILL EXISTING TAPE MOTION
0576  6766            DTCX            /NOW
0577  5526            JMP I STX       /RETURN TO START
```

```
0600            *600
                /WAIT FOR THE DECTAPE FLAG TO RISE

0600  0000  STALL, 0
0601  7200        CLA
0602  6772        DTRB            /READ TCU "B" REGISTER
0603  7510        SPA             /ERROR?
0604  5242        JMP   ERROR     /YES, DECIDE WHAT TO DO
0605  7010        RAR             /DECTAPE FLAG TO LINK
0606  7620        SNL   CLA       /FLAG?
0607  5202        JMP   .-5       /NO: CONTINUE WATCH
0610  6764  RERR, DTXA            /RESET THE DECTAPE FLAG
0611  3031        DCA   ERX       /CLEAR THE END TAPE FLAG
0612  5600        JMP I STALL     /GOT FLAG, EXIT

                /DRIVE TAPE INTO THE END ZONE, AND TURN IT
                /AROUND.
                /IF C(AC)=0400, TAPE INTO REVERSE END ZONE
                /IF C(AC)=0000, TAPE INTO FORWARD END ZONE

0613  0000  TRN,  0
0614  2031        ISZ   ERX       /END ZONE IS LEGAL
0615  3023        DCA   W4        /SAVE DIRECTION
0616  1160        TAD   DT0200    /MOVE FUNCTION,GO
0617  1023        TAD   W4        /DIRECTION TO MOVE
0620  1030        TAD   DTA       /DRIVE TO MOVE
0621  6766        DTCX            /CLEAR AND RESET "A"
0622  4533        JMS I WAIT      /FOR END ZONE FLAG
0623  1165        TAD   DT0610    /SEARCH, GO
0624  1023        TAD   W4        /DIRECTION TO SEARCH
0625  0232        AND   C0777     /DELETE OVERFLOW BIT
0626  1030        TAD   DTA       /SET THE DECTAPE
0627  6766        DTCX            /RESET STATUS "A"
0630  3031        DCA   ERX       /END ZONE NOT LEGAL NOW
0631  5613        JMP I TRN       /RETURN TO SEQUENCE
0632  0777  C0777, 0777

                /WAIT TILL WORD COUNT REGISTER GOES TO ZERO

0633  0000  MINI,  0000
0634  7300        CLA   CLL
0635  1534        TAD I WC        /WORD COUNT TO THE AC
0636  7640        SZA   CLA       /WORD COUNT EQUAL TO ZERO?
0637  5235        JMP   .-2       /NO: BE PATIENT
0640  6764        DTXA            /YES: RESET THE DECTAPE FLAG
0641  5633        JMP I MINI      /RETURN TO SEQUENCE
```

```
                        /AN ERROR FLAG HAS BEEN SET. IN SOME CASES
                        /END ZONE IS LEGAL, OTHERWISE, A RESTART ATTEMPT
                        /MAY BE INITIATED.

                        /DETERMINE WHICH FLAG SET THE DECTAPE FLAG

0642  3024    ERROR,    DCA    W5          /SAVE "B" REGISTER
0643  1030              TAD    DTA         /GOING TO KILL
0644  6766              DTCX               /TAPE MOTION
0645  1024              TAD    W5          /RESTORE "B" REGISTER
0646  7006              RTL                /POSITION BITS 1+2
0647  7510              SPA                /END OF TAPE FLAG?
0650  5261              JMP    ZEOT        /YES; GO TO ROUTINE
0651  7430              SZL                /MARK TRACK ERROR?
0652  5274              JMP    ZMKTK       /YES; GO TO ROUTINE
0653  7006              RTL                /POSITION BITS 2+3
0654  7510              SPA                /PARITY ERROR?
0655  5304              JMP    ZPAR        /YES; GO TO PARITY ERROR ROUTINE
0656  7630              SZL    CLA         /SELECT ERROR?
0657  5312              JMP    ZSEL        /YES; GO TO ROUTINE
0660  5320              JMP    ZTIM        /MUST BE TIMING ERROR

                        /END OF TAPE FLAG FOUND, SEE IF IT'S LEGAL

0661  7300    ZEOT,     CLA    CLL         /CLEAR REMAINS
0662  1031              TAD    ERX         /SWITCH
0663  7640              SZA    CLA         /ERROR?
0664  5210              JMP    RERR        /OK, IT'S LEGAL

                        /NOT LEGAL END ZONE FLAG

0665  4531              JMS I  TYPE
0666  0516              0516    /EN
0667  0440              0440    /D
0670  2401              2401    /TA
0671  2005              2005    /PE
0672  4000              4000    / 0
0673  5325              JMP    ZCOM

                        /MARK TRACK ERROR

0674  4531    ZMKTK,    JMS I  TYPE
0675  1501              1501    /MA
0676  2213              2213    /RK
0677  4024              4024    / T
0700  2201              2201    /RA
0701  0313              0313    /CK
0702  4000              4000    / 0
0703  5325              JMP    ZCOM
```

```
                        /PARITY ERROR

0704  4531    ZPAR,   JMS I TYPE
0705  2001            2001   /PA
0706  2211            2211   /RI
0707  2431            2431   /TY
0710  4000            4000   / 0
0711  5325            JMP    ZCOM

                        /SELECT ERROR

0712  4531    ZSEL,   JMS I TYPE
0713  2305            2305   /SE
0714  1405            1405   /LE
0715  0324            0324   /CT
0716  4000            4000   / 0
0717  5325            JMP    ZCOM

                        /TIMING ERROR

0720  4531    ZTIM,   JMS I TYPE
0721  2411            2411   /TI
0722  1511            1511   /MI
0723  1607            1607   /NG
0724  4000            4000   / 0

                        /TYPE "ERROR PHASE X"

0725  1032    ZCOM,   TAD    PHASE     /WHAT PHASE OF OPERATION
0726  1373            TAD    PFORM     /WAS THE MACHINE IN
0727  3337            DCA    TFORM     /WHEN ERROR OCCURED
0730  4531            JMS I TYPE
0731  0522            0522   /ER
0732  2217            2217   /RO
0733  2240            2240   /R
0734  2010            2010   /PH
0735  0123            0123   /AS
0736  0540            0540   /E
0737  4060    TFORM,  4060   / X
0740  4345            4345   /CR+LF
0741  0000            0000   /END
0742  4532            JMS I TYPIN

                        /HE CAN RESTART IF HE TYPES "RETRY"

0743  4513            JMS I COMPAR
0744  0322            0322  /R
0745  0305            0305  /E
0746  0324            0324  /T
0747  0322            0322  /R
0750  0331            0331  /Y
0751  0000            0000  /0
0752  5515            JMP I IT        /GUESS HE DOESN'T WISH TO TRY AGAIN
```

```
                        /ATTEMPT RESTART, NOTE, "ATTEMPT"

0753  1032              TAD   PHASE     /RESTART ACCORDING TO
0754  1357              TAD   ZFORM     /WHICH PHASE WAS HE IN
0755  3360              DCA   .+3
0756  5760              JMP I .+2
0757  0761  ZFORM,      .+2
0760  0000              0000
0761  5766              JMP I .+5       /PHASE 0
0762  5767              JMP I .+5       /PHASE 1
0763  5770              JMP I .+5       /PHASE 2
0764  5771              JMP I .+5       /PHASE 3
0765  5772              JMP I .+5       /PHASE 4
0766  1000              START
0767  1622              PSER
0770  2000              DOBLK
0771  2200              DBN
0772  7000              NOP
0773  4060  PFORM,      4060
```

```
/HERE STARTS THIS PROGRAM. IT WILL ASK THE
/OPERATOR FOR  DRIVE NUMBERS, THEN ASK HIM FOR
/A DIRECTION ON WHAT TO DO WITH THE DRIVES.

/THE SEQUENCE FOR MARKING A TAPE WOULD APPEAR AS:


/DTA? (3 OR 1 2 3 OR 2 4 7)
/DIRECT? (MARK 1215)
/2277 WORDS, 0256 BLOCKS.OK? YES OR NO
/(YES)


/THAT DATA IN PARENTHESIS IS TYPED BY THE OPERATOR
/(HE DOESN'T TYPE THE PARENTHESIS)
/IF HE HAD ANSWERED NO, "DIRECT?" WOULD BE TYPED OUT.
/IF THE DRIVE WAS WRONG, HE WOULD TYPE RESTART.
/IF HE HAD TYPED "MARK" IN RESPONSE TO "DIRECT?" THE
/TAPE WOULD BE MARKED WITH THE STANDARD PDP-8 CONFIGURATION.
/IF HE HAD TYPED "MARK 384" THE TAPE WOULD
/BE MARKED WITH THE STANDARD PDP-10 CONFIGURATION
/NOTE: THE WORD AND BLOCK NUMBERS ARE TYPED IN OCTAL
/IF A MISTAKE OCCURS ON THE OPERATORS PART (WITH REFERANCE
/TO BLOCK * WORD SIZE) HE WILL BE TOLD ABOUT IT
```

```
                1000            *1000

                                /MAKE A CALL FOR THE DECTAPE NUMBERS TO BE
                                /WORKED.

1000  4531      START,  JMS I TYPE      /SET UP TYPER
1001  4543              4543  /CR+LF
1002  4300              4300  /LF+END
1003  4531      TYQU,   JMS I TYPE      /"DTA?"
1004  0424              0424  /DT
1005  0177              0177  /A?
1006  4000              4000  / END

                                /WAIT FOR A REPLY

1007  4532              JMS I TYPIN     /GET NUMBERS
1010  1110              TAD     BADD    /INITIALIZE POINTER (BFR)
1011  7001              IAC             /(BADD=BUFFER-1, SO BUMP THE AC)
1012  3111              DCA     BFR     /TO START OF INPUT BUFFER
1013  3374              DCA     DCTR    /INITIALIZE DTA COUNTER TO 0
1014  3346              DCA     CRFLAG  /CLEAR FLAG SO CR NOT ACCEPTIBLE
1015  1072      CRCHK,  TAD     CRCOD   /GET CODE FOR CAR. RETN
1016  7041              CIA             /NEGATE IT
1017  1511              TAD I   BFR     /SEE IF NEXT CHAR. IN
1020  7450              SNA             /BUFFER IS CAR. RETN.
1021  5244              JMP     OKCR    /YES: SEE IF C.R. LEGAL HERE
1022  3346              DCA     CRFLAG  /NO: SO C.R. IS LEGAL NOW
1023  1055      VALCHK, TAD     C261    /SEE IF # IS LESS THAN
1024  7041              CIA             /ASCII 1 (261)
1025  1511              TAD I BFR       /SUBTRACT BUFFER DATA
1026  7710              SPA   CLA       /IS IT LESS THAN ASII 0?
1027  5203              JMP   TYQU      /YES: TELL OUTSIDE WORLD
1030  1057              TAD   C270      /NO: SEE IF GREATER THAN
1031  7040              CMA             /ASC II 8 (270)
1032  1511              TAD I BFR       /SUBTRACT BUFFER DATA
1033  7700              SMA   CLA       /GREATER THAN ASCII 7?
1034  5203              JMP   TYQU      /YES: TELL OUTSIDE WORLD
1035  1511              TAD I BFR       /NO: ACCEPT BUFFER
1036  7012              RTR
1037  7012              RTR             /4 BITS RIGHT
1040  0064              AND   C7000     /ISOLATE DTA
1041  4347              JMS     REPEAT  /GO CHECK FOR REPEATED DTA AND STORE #
1042  2111              ISZ     BFR     /INCREMENT INPUT BUF. PTR.
1043  5215              JMP     CRCHK   /GO LOOK AT NEXT CHAR.

                                /THIS SECTION CHECKS TO SEE IF THERE HAS BEEN ANY
                                /VALID INPUT ONCE A CARRIAGE RETURN IS SEEN
1044  7200      OKCR,   CLA             /CLEAR AC
1045  1346              TAD     CRFLAG  /LOAD CR FLAG: 0 MEANS NO GOOD
1046  7650              SNA     CLA
1047  5200              JMP     START   /0: NO VALID INPUT; RESTART
1050  1374              TAD     DCTR    /NOT 0: SO HAVE VALID INPUT
1051  1376              TAD     DBUFAD  /CALCULATE END OF DTA LIST +1
1052  3375              DCA     DBUFPT  /STORE IT IN BUFFER POINTER, THEN
1053  7040              CMA             /COMPLEMENT THE AC AND
```

```
1054  3775          DCA I   DBUFPT  /TERMINATE DTA LIST WITH 7777
1055  7200  INIT1,  CLA             /CLEAR AC IF COME THRU LOC IT
1056  1376          TAD     DBUFAD  /AND RESET LIST POINTER
1057  3375          DCA     DBUFPT  /TO START OF LIST
1060  4745          JMS I   GETDTA  /GO GET A DTA NUMBER

                    /INFORM THE OPERATOR THAT THE PROGRAM IS SET TO START
                    /TYPE "DIRECT" AND WAIT FOR THE REPLY

1061  4531  INIT,   JMS I TYPE      /MESSAGE OUT
1062  0411          0411  /DI
1063  2205          2205  /RE
1064  0324          0324  /CT
1065  7740          7740  /?
1066  0000          0000  /END
1067  4532          JMS I TYPIN     /WAIT FOR A REPLY
1070  4513          JMS I COMPAR    /DID HE TYPE "MARK"?
1071  0315          0315  /M
1072  0301          0301  /A
1073  0322          0322  /R
1074  0313          0313  /K
1075  0000          0000  /END
1076  5301          JMP   .+3
1077  5700          JMP I .+1
1100  1200          MARK            /TO MARK A TAPE
```

```
                        /SEE IF HE TYPED "RDR" (READ AND TYPE FIRST 12
                        /BLOCK NUMBERS IN REVERSE).

1101  4513              JMS I COMPAR
1102  0322              0322  /R
1103  0304              0304  /D
1104  0322              0322  /R
1105  0000              0000  /0
1106  5311              JMP   .+3
1107  5710              JMP I .+1
1110  2450              RDR             /TYPE BLOCKS

                        /SEE IF HE TYPED "RDF" (READ AND TYPE FIRST 12
                        /BLOCK NUMBERS FORWARD).

1111  4513              JMS I COMPAR
1112  0322              0322  /R
1113  0304              0304  /D
1114  0306              0306  /F
1115  0000              0000  /0
1116  5321              JMP   .+3
1117  5720              JMP I .+1
1120  2400              RDFA            /TYPE BLOCKS

                        /SEE IF HE TYPED "SAME" (MEANING MARK A TAPE
                        /USING THE SAME CONSTANTS AS BEFORE).

1121  4513              JMS I COMPAR
1122  0323              0323  /S
1123  0301              0301  /A
1124  0315              0315  /M
1125  0305              0305  /E
1126  0000              0000  /0
1127  5332              JMP   .+3
1130  5731              JMP I .+1
1131  1353              RSTSM           /TO MARK AS BEFORE

                        /SEE IF HE TYPED "RESTART"

1132  4513              JMS I COMPAR
1133  0322              0322  /R
1134  0305              0305  /E
1135  0323              0323  /S
1136  0324              0324  /T
1137  0301              0301  /A
1140  0322              0322  /R
1141  0324              0324  /T
1142  0000              0000  /0
1143  4143              JMS   QU        /MUST BE NONSENSE
1144  5200              JMP     START   /START ALL OVER
1145  1703  GETDTA,     NUDTA           /POINTER TO ROUTINE TO SWITCH UNITS
1146  0000  CRFLAG,     0               /=0, CR NO GOOD; NOT 0, CR IS OK
```

```
                        /SUBROUTINE TO CHECK FOR REPEATED DTA NUMBERS
                        /DTA # TO COMPARE TO LIST IS IN AC ON ENTRY--THIS
                        /ROUTINE STORES THE DTA # IF IT IS NEW AND IGNORES IT
                        /IF IT IS NOT-CALL BY JMS REPEAT WITH DTA # IN AC
1147  0000              REPEAT, 0
1150  3377                      DCA     DNUM    /TEM STORAGE FOR NEW DTA #
1151  1376                      TAD     DBUFAD  /INITIALIZE POINTER (DBUFPT)
1152  3375                      DCA     DBUFPT  /TO START OF DTA LIST
1153  1374                      TAD     DCTR    /LOAD NUM. OF DTAS STORED
1154  7040                      CMA             /COMPLEMENT IT
1155  3373                      DCA     COMCTR  /STORE IN COMPARE COUNTER
1156  2373              COMCHK, ISZ     COMCTR  /DONE WITH ALL COMPARES?
1157  5364                      JMP     DOCOMP  /NO: GO DO COMPARE
1160  1377                      TAD     DNUM    /YES: STORE NEW DTA#
1161  3775                      DCA I   DBUFPT  /AT END OF LIST
1162  2374                      ISZ     DCTR    /INCR. # OF DTAS STORED
1163  5747                      JMP I   REPEAT  /RETURN

                        /THIS SECTION DOES THE ACTUAL COMPARISON BETWEEN
                        /THE DTA# PASSED TO THE ROUTINE AND A NUMBER ON THE LIST

1164  1775              DOCOMP, TAD I   DBUFPT  /GET NEXT DTA NUMBER FROM LIST
1165  7041                      CIA             /NEGATE IT
1166  1377                      TAD     DNUM    /ADD IN DTA NUMBER PASSED
1167  7650                      SNA     CLA     /ARE THEY THE SAME?
1170  5747                      JMP I   REPEAT  /YES: RETURN
1171  2375                      ISZ     DBUFPT  /NO: INCREMENT LIST POINTER
1172  5356                      JMP     COMCHK  /SEE IF DONE ALL COMPARES
                        /
                        /
1173  0000              COMCTR, 0               /COUNTER FOR # OF LIST COMPARISONS TO BE DONE
1174  0000              DCTR,   0               /COUNTER FOR # OF DTAS IN LIST
1175  0000              DBUFPT, 0               /POINTER TO CURRENT POSITION IN DTA LIST
1176  1725              DBUFAD, DTABUF          /START OF DTA NUM. LIST
1177  0000              DNUM,   0               /TEM STORAGE FOR DTA #
                        /
                        /
                        PAUSE
```

```
                1200            *1200
                                /MARK WAS TYPED IN, IF W1-1 IS NOT A "K",ASSUME THAT
                                /A NUMBER WAS TYPED IN, AND VERIFY THIS. IF W1-1 IS
                                /A "K", ASSUME STANDARD FORMAT.(W1=LAST ENTRY INTO THE BUFFER)

        1200    1171    MARK,   TAD     BINCON      /ADDRESS OF FIRST BINARY
        1201    3024            DCA     W5          /CONSTANT FOR DEC TO BIN
        1202    3033            DCA     TOTAL       /WILL BE BINARY EQUIVILANT

                                /SAVE C(X1) FOR DECREMENT THROUGH BUFFER

        1203    7240    DNC,    CLA     CMA         /DECREMENT BUFFER ADDRESS
        1204    1020            TAD     W1          /ADDRESS BY 1
        1205    3020            DCA     W1          /W1=SWEEP ADDRESS

                                /LOOK FOR END OF PROCESSING BY LOOKING FOR A "K" IN BUFFER

        1206    1073            TAD     LETK        /LETTER ASCII "K"
        1207    7041            CIA                 /SUBTRACT FROM CHARACTER
        1210    1420            TAD I   W1          /IN BUFFER
        1211    7650            SNA     CLA         /EQUAL?
        1212    5244            JMP     DIV3        /YES; SEE IF DIVISIBLE BY 3

                                /VERIFY THIS CHARACTER AS BEING OF DECIMAL ORIGIN

        1213    1054            TAD     C260        /ASCII FOR 0
        1214    7041            CIA                 /TO SEE IF CHARACTER
        1215    1420            TAD I   W1          /IS LESS THAN 260
        1216    7710            SPA     CLA         /IS IT?
        1217    5520            JMP I   QU1         /YES; NOT DECIMAL CHARACTER
        1220    1060            TAD     C271        /ASCII FOR 9
        1221    7040            CMA                 /TO SEE IF GREATER THAN
        1222    1420            TAD I   W1          /9
        1223    7700            SMA     CLA         /IS IT?
        1224    5520            JMP I   QU1         /NOT A DECIMAL CHARACTER
```

```
                        /CHARACTER IS DECIMAL. NOW CONVERT IT TO BINARY
                        /REMEMBER POSITION OF CHARACTER IN BUFFER MAY BE
                        /10,100,1000.

1225  1420                      TAD I W1        /ISOLATE THE NUMBER
1226  0042                      AND   C0017     /FOR PROPER CONVERSION
1227  7450                      SNA             /IF 0, NO BINARY CONVERSION NEEDED
1230  5242                      JMP   IBS       /YES: 0: INCREMENT BINARY CONVERSION

                        /NOT 0, SET UP CONVERSION LOOP

1231  7141                      CLL   CIA       /NUMBER OF ADDITIONS
1232  3023                      DCA   W4        /TO NEGATIVE FOR ISZ
1233  1424                      TAD I W5        /BINARY POSITION TO C(ACC)
1234  1033                      TAD   TOTAL     /ADD TO PRESENT TOTAL
1235  7430                      SZL             /CHECK ON TO MANY WORDS
1236  5521                      JMP I QU2       /TO MANY WORDS CALLED FOR
1237  3033                      DCA   TOTAL     /KEEP RUNNING SUM
1240  2023                      ISZ   W4        /LAST ADDITION?
1241  5233                      JMP   .-6       /NO: ADD AGAIN

                        /FINAL ADDITION FOR THIS POSITION COMPLETED

1242  2024              IBS,    ISZ   W5        /NEXT POSITION
1243  5203                      JMP   DNC       /DO NEXT CHARACTER

                        /LAST CHARACTER COMPLETED. SEE IF DIVISIBLE BY 3
                        /IF NOT A NORMAL INPUT

1244  1033              DIV3,   TAD   TOTAL     /GET TOTAL WORDS
1245  7450                      SNA             /IF TOTAL 0, NORMAL INPUT
1246  1052                      TAD   C201      /129 OCT. THIS TEST REDUNDANT
1247  1042                      TAD   C0017     /ADD CONSTANT 15 TO TOTAL
1250  3033                      DCA   TOTAL     /FOR FUTURE CONSIDERATIONS
1251  3034                      DCA   VAR1      /# OF WORDS/3 FOR MARK TRACK WRITING
1252  1033                      TAD   TOTAL     /RESTORE IN THE ACC
1253  7100                      CLL             /TO DIVIDE BY 3, LINK KEEPS OVERFLOW
1254  1076                      TAD   M3        /SUBTRACT 3
1255  2034                      ISZ   VAR1      /ON EACH DIVISION, KEEP RUNNING SUM
1256  7440                      SZA             /IF AC = 0,NO REMAINDER
1257  7420                      SNL             /WHEN LINC GOES TO 0, DIVISION ENDED
1260  7410                      SKP             /NOW SEE IF IT DIVIDED EVENLY
1261  5253                      JMP   .-6       /SUBTRACT 3 MORE
1262  7640                      SZA   CLA       /IF 0,OK, OTHERWISE ERROR
1263  5523                      JMP I QU4       /NOT DIVISIBLE BY 3

                        /CORRECT "VAR1" ( THE NUMBER OF WORDS/3) FOR THE +15
                        /ADDED JUST ABOVE AND AN INHERANT +2 DUE TO MARK TRACK
                        /CONFIGURATION TO BE WRITTEN.

1264  1101                      TAD   M7        /SUBTRACT 7 FROM PHONY SETUP
1265  1034                      TAD   VAR1      /GIVING THE NUMBER OF TIMES
1266  7041                      CIA             /TO BE USED LATER IN A ISZ
1267  3034                      DCA   VAR1      /DATA MARK WILL BE WRITTEN
```

```
                        /COMPUTE A VALUE FOR TOTAL NUMBER OF BLOCKS
                        /RECORD SIZE * 15 INTO 636160 OCT.

1270  1066              TAD   C7714     /EXTENDED 64 VALUE, SETS AC#2
1271  3020              DCA   W1        /SET FOR 640000
1272  4775              JMS I FORM10    /PATCH TO CHECK FOR STD.10 FORMAT
1273  1063              TAD   C1620     /VERNIER ADJUSTMENT FOR FORMULA
1274  7100              CLL             /ACC#2 CARRY FUNCTION
1275  1033              TAD   TOTAL     /WORD COUNT
1276  2026              ISZ   BLOCKS    /+1 TO BLOCK COUNT
1277  7410              SKP
1300  5522              JMP I QU3       /TO MANY BLOCKS CALLED FOR
1301  7420              SNL             /CARRY INTO ACC#2?
1302  5275              JMP   .-5       /NO: CONTINUE COUNT
1303  2020              ISZ   W1        /YES: FULLY DIVIDED?
1304  5274              JMP   .-10      /NO: CONTINUE PROCESS
1305  7300              CLA   CLL       /C(ACC)+ C(L)=0
1306  1026  F10RTN, TAD  BLOCKS        /FOR MARK TRACK (COME HERE FR F10PAT IF 10 FRMT)
1307  7040              CMA             /WRITING
1310  3035              DCA   VAR2      /SEE MARK WRITE

                        /VALUES FOR BLOCK AND RECORD SIZE HAVE BEEN
                        /COMPUTED. TELL OUTSIDE WORLD AND GET THE OK.

1311  1033              TAD   TOTAL     /SUBTRACT 15 FROM TOTAL
1312  1067              TAD   C7761     /WORDS FOOLING OPERATOR
1313  3033              DCA   TOTAL     /CORRECTED FOR TAPE WRITING
1314  1033              TAD   TOTAL     /FOR OCTAL TYPEOUT
1315  4530              JMS I TYOCT     /TYPE OCTAL WORDS
1316  4531              JMS I TYPE      /TYPE MESSAGE
1317  4027              4027  / W
1320  1722              1722  /OR
1321  0423              0423  /DS
1322  5400              5400  /, END
1323  1026              TAD   BLOCKS    /TYPE OUT BLOCK #S
1324  7001              IAC             /TO FOOL THE OPERATOR
1325  4530              JMS I TYOCT     /IN OCTAL
1326  4531              JMS I TYPE      /TYPE MESSAGES
1327  4002              4002  / B
1330  1417              1417  /LO
1331  0313              0313  /CK
1332  2356              2356  /S,
1333  1713              1713  /OK
1334  7733              7733  /?(
1335  3105              3105  /YE
1336  2340              2340  /S
1337  1722              1722  /OR
1340  4016              4016  / N
1341  1735              1735  /O)
1342  4543              4543  /CR+LF
1343  0000              0000  /END
1344  4532              JMS I TYPIN     /WAIT FOR REPLY
```

```
                        /SEE IF A YES OR NO ANSWER WAS GIVEN

1345  4513              JMS I COMPAR
1346  0331              0331   /Y
1347  0305              0305   /E
1350  0323              0323   /S
1351  0000              0000   /END
1352  5515              JMP I IT

                        /SEE IF THE DRIVE IS OK

1353  1153  RSTSM,      TAD    DT0060     /GIVE WRTM, NO GO
1354  1030              TAD    DTA        /AND DTA #
1355  6766              DTCX              /ORDER EXECUTE
1356  3020              DCA    W1         /STALL FUNCTION
1357  6772  CDTRD,      DTRB              /READ STATUS "B"
1360  7700              SMA    CLA        /ERROR?
1361  5371              JMP    CIZ        /NO: TIME OUT STALL
1362  4531              JMS I TYPE        /YES: INCORRECT SETUP
1363  2305              2305   /SE
1364  2425              2425   /TU
1365  2077              2077   /P
1366  0000              0000   /END
1367  5770              JMP I .+1
1370  1000              START

                        /STALL FOR A WHILE FOR THE INTERRUPT

1371  2020  CIZ,        ISZ    W1         /ONE ROUND'S WORTH
1372  5357              JMP    CDTRD      /OF ISZ
1373  5774              JMP I .+1
1374  1400              STMK              /OK, GO DO THE MARK TRACK
1375  1560  FORM10,     F10PAT
```

```
                1400            *1400
                                /SET THE TAPE INTO MOTION. ALL VARIABLES ARE
                                /SET. FROM THIS POINT ON, CONTROL IS EXECUTED
                                /VIA THE WCO INTERRUPT

                                /CLEAR OUT STATUS "A" AND RELOAD IT WITH CONTINUOUS
                                /WRITE TIMING AND MARK TRACK COMMAND

        1400    1162    STMK,   TAD     DT0360      /FWD, CONT, T*M,GO,INT
        1401    1030            TAD     DTA         /ADD IN THE DTA
        1402    6766            DTCX                /CLEAR FLAGS START MOTION
        1403    3032            DCA     PHASE       /FOR ERROR ROUTINE
        1404    1035            TAD     VAR2        /TO MAKE A RESTART FOR THE "SAME"
        1405    3025            DCA     W6          /OPTION  POSSIBLE

                                /WRITE END ZONE, WRITE ABOUT 10' OF THIS
                                /CONFIGURATION, 4044
                                /                   0440    ON TAPE AS
                                /                   4404    (5555) OCTAL.

        1406    3020            DCA     W1          /CLEAR COUNTER, 7777= ABOUT 10'
        1407    1312    CEZ,    TAD     REZ         /LOAD ADDRESS OF DATA
        1410    3512            DCA I   CA          /TO BE WRITTEN INTO THE CA
        1411    1076            TAD     M3          /LOAD # WORDS TO BE WRITTEN INTO
        1412    3534            DCA I   WC          /WC LOCATION

                                /WAIT FOR INTERRUPT, TEST FOR END OF
                                /END ZONE WRITING.

        1413    4533            JMS I   WAIT        /FOR INTERRUPT
        1414    2020            ISZ     W1          /END OF FOOTAGE?
        1415    5207            JMP     CEZ         /NOT END FOOTAGE, CONTINUE
                                                    /OK, WRITE INTERBLOCK SYNC

                                /WRITE INTERBLOCK SYNC, SINCE THIS CONFIGURATION
                                /ACT AS A NOP TO THE TCU, AT THE BEGINING OF
                                /TAPE, MORE LENGTH OF THIS IS NEEDED FOR TURN AROUND
                                /TIME TO GUARANTEE BLOCK 0000 TO THE LIBRARY SYSTEM
                                /THEREFORE AT THE BEGINING OF TAPE ONLY, WRITE SEVERAL
                                /INTERBLOCK ZONES

        1416    1103            TAD     M144        /NUMBER OF TIMES TO
        1417    3020            DCA     W1          /WRITE INTERBLOCK SYNC
        1420    4224            JMS     INBLSY      /WRITE 1 INTERBLOCK SYNC
        1421    2020            ISZ     W1          /CONFIGURATION, TEST END
        1422    5220            JMP     .-2         /NOT TOTAL FOOTAGE, WRITE AGAIN
        1423    5235            JMP     WDZ         /COMPLETED, GO ON
```

```
                        /AT NORMAL RETURN, WRITE ONLY ONE INTERBLOCK SYNC
                        /CONFIGURATION, APPEARS AS      0404
                        /                               0404   ON TAPE AS
                        /                               0404   2525 OCTAL

1424  0000      INBLSY, 0
1425  1316              TAD   IBZ        /COUNTER AND WORD
1426  3512              DCA I CA         /COUNT WITH KEYS
1427  1076              TAD   M3         /FOR CONTROL
1430  3534              DCA I WC
1431  1034              TAD   VAR1       /RESET THE WORDS
1432  3024              DCA   W5         /PER BLOCK COUNTER

                        /WAIT FOR INTERRUPT, RETURN TO SEQUENCE

1433  4533              JMS I WAIT       /FOR INTERRUPT
1434  5624              JMP I INBLSY


                        /WRITE FORWARD BLOCK MARK AND REVERSE GUARD
                        /THREE WORDS    0404
                        /               4004   ON TAPE AS
                        /               4040   2632 OCTAL

1435  1322      WDZ,    TAD   FBM        /ADDRESS OF PATTERN
1436  3512              DCA I CA         /TO CURRENT ADDRESS
1437  1076              TAD   M3         /NUMBER OF WORDS
1440  3534              DCA I WC         /TO WORD COUNTER
1441  4533              JMS I WAIT       /DROP THROUGH AFTER WRITE


                        /WRITE LOCK MARK, REVERSE CKSUM, REVERSE FINAL,REV PREFINAL
                        /SIX WORDS      1. 0040   4. 0040
                        /               2. 0000   5. 0000   ON TAPE OCTAL
                        /               3. 4000   6. 4000   10101010

1442  1326              TAD   WLMRF      /ADDRESS OF PATTERN
1443  3512              DCA I CA         /TO CURRENT ADDRESS
1444  1077              TAD   M6         /NUMBER OF WORDS
1445  3534              DCA I WC         /TO WORD COUNTER
1446  4533              JMS I WAIT       /DROP THROUGH AFTER WRITE


                        /   WRITE THE DATA TRACK. SINCE THE LENGTH OF EACH
                        /RECORD IS A VARIABLE, "VAR1" KEEPS TRACK OF THE
                        / UMBER OF TIMES THIS CONFIGURATION WILL BE WRITTEN
                        /"VAR1" WAS DECIDED FROM ABOVE IN THE FORMULA
                        / RANSLATION SECTION
                        /THREE WORDS    4440
                        /               0044   ON TAPE AS
                        /               4000   7070 OCTAL
1447  1335      DTRK,   TAD   DZ         /LOAD ADDRESS OF THE DATA
1450  3512              DCA I CA         /CONFIGURATION INTO CA
1451  1076              TAD   M3         /LOAD # WORDS
1452  3534              DCA I WC         /INTO WORD COUNT
```

```
                        /WRITE ONE SET TEST "VAR1" FOR LAST SET

1453  4533                      JMS I WAIT      /ONE CONFIGURATION
1454  2024                      ISZ   W5        /LAST?
1455  5247                      JMP   DTRK      /NOW WRITE DATA MARK TRACK AGAIN

                        /   MARK TRACK CODE FOR DATA IS COMPLETE, NOW WRITE
                        /PREFINAL, FINAL, CHECKSUM AND REVERSE CHECKSUM.
                        /SIX WORDS      1 4440   4 4440
                        /               2 4444   5 4444   ON TAPE AS
                        /               3 4044   6 4044   73737373 OCTAL

1456  1341                      TAD   FEZ       /LOAD ADDRESS OF
1457  3512                      DCA I CA        /DATA CONFIGURATION INTO CA
1460  1100                      TAD   M6        /LOAD # WORDS
1461  3534                      DCA I WC        /INTO WORD COUNT
1462  4533                      JMS I WAIT      /TILL COMPLETED WRITE


                        /WRITE GUARD, REVERSE BLOCK
                        /THREE WORDS    4040
                        /               0440   ON TAPE AS
                        /               0404   5145 OCTAL

1463  1350                      TAD   GRZ       /DATA ADDRESS TO
1464  3512                      DCA I CA        /THE CA
1465  1076                      TAD   M3        /NUMBER OF WORDS
1466  3534                      DCA I WC        /TO WORD COUNT
1467  4533                      JMS I WAIT      /TILL COMPLETE


                        /THIS COMPLETE SET OF DATA TRANSFERES
                        /COMPLETES ONE BLOCK ON TAPE, SINCE THE
                        /NUMBER OF BLOCKS IS VARIABLE, "VAR2" IS
                        /USED TO RECYCLE, "VAR2" WAS SET UP ABOVE IN
                        /THE FORMULA TRANSLATION SECTION

1470  4224                      JMS   INBLSY    /WRITE INTERBLOCK SYNC
1471  2025                      ISZ   W6        /TOTAL NUMBER OF BLOCKS
1472  5235                      JMP   WDZ       /WRITTEN? NO:
```

```
                        /ALL DATA BLOCKS HAVE BEEN WRITTEN,
                        /NOW PROVIDE A BUFFER ZONE OF INTERBLOCK SYNC AT THE END
                        /OF TAPE AS AT THE START OF TAPE

1473  1103                      TAD   M144       /ABOUT TWO BLOCKS(STANDARD) WORTH
1474  3020                      DCA   W1         /ABOUT 100 TIMES
1475  4224                      JMS   INBLSY     /WRITE ONE PATTERN
1476  2020                      ISZ   W1         /AT END YET?
1477  5275                      JMP   .-2        /NO CONTINUE WRITING INTERBLOCK SYNC

                        /COMPLETED BLOCK WRITING
                        /WRITE ANOTHER 10' OF END ZONE (FORWARD)
                        /BEFORE LOADING BLOCK NUMBERS.
                        /THREE WORDS    0400
                        /               4004   ON TAPE AS
                        /               0040   2222 OCTAL

1500  3020                      DCA   W1         /ISZ*10 FEET
1501  1354              WE2F,   TAD   E2M        /LOAD ADDRESS OF DATA
1502  3512                      DCA I CA         /INTO CA
1503  1076                      TAD   M3         /NUMBER OF WORDS
1504  3534                      DCA I WC         /WORD COUNT

                        /WRITE 1 SET, CHECK END OF 10',

1505  4533                      JMS I WAIT       /TILL COMPLETE
1506  2020                      ISZ   W1         /END OF FOOTAGE?
1507  5301                      JMP   WE2F       /NO, CONTINUE WITH END ZONE
1510  5711                      JMP I .+1        /GO AND START BLOCK NUMBER
1511  1600                      MWTM             /SEQUENCING
```

```
                        /THESE ARE THE DATA CONFIGURATIONS FOR THE MARK TRACK


                        /REVERSE END ZONE

1512  1512    RE2,    .
1513  4044            4044    /ON TAPE AS 5555 (OCT)
1514  0440            0440
1515  4404            4404

                        /INTERBLOCK SYNC

1516  1516    IB2,    .
1517  0404            0404    /ON TAPE AS 2525 (OCT)
1520  0404            0404
1521  0404            0404

                        /FORWARD BLOCK MARK AND REVERSE GUARD

1522  1522    FBM,    .
1523  0404            0404    /ON TAPE AS 2632 (OCT)
1524  4004            4004
1525  4040            4040

                        /LOCK MARK, REVERSE CHECKSUM, REVERSE FINAL
                        /AND REVERSE PREFINAL

1526  1526    WLMRF,  .
1527  0040            0040    /ON TAPE AS 10101010 (OCT)
1530  0000            0000
1531  4000            4000
1532  0040            0040
1533  0000            0000
1534  4000            4000

                        /DATA MARK

1535  1535    D2,     .
1536  4440            4440    /ON TAPE AS 7070 (OCT)
1537  0044            0044
1540  4000            4000

                        /PREFINAL, FINAL, FWD CHECKSUM, AND REVERSE LOCK

1541  1541    FE2,    .
1542  4440            4440    /ON TAPE AS 73737373 (OCT)
1543  4444            4444
1544  4044            4044
1545  4440            4440
1546  4444            4444
1547  4044            4044
```

```
                        /FORWARD GUARD AND REVERSE BLOCK NUMBER

1550  1550              GRZ,    .
1551  4040                      4040    /ON TAPE AS 5145 (OCT)
1552  0440                      0440
1553  0404                      0404

                        /FORWARD END ZONE

1554  1554              EZM,    .
1555  0400                      0400    /ON TAPE AS 2222 (OCT)
1556  4004                      4004
1557  0040                      0040
                        /SUBROUTINE TO SEE IF USER TYPED MARK 384
                        /TO SPECIFY STANDARD PDP-10 FORMAT
1560  0000              F10PAT, 0
1561  3026                      DCA     BLOCKS  /CLEAR LOC. BLOCKS IN CASE NOT 10-FORMAT
1562  1033                      TAD     TOTAL   /AND GET NUMBER TYPED BY USER
1563  1373                      TAD     M617    /WAS IT 3847
1564  7640                      SZA     CLA
1565  5760                      JMP I   F10PAT  /NO-RETURN
1566  3020                      DCA     W1      /YES-CLEAR W1 FOR WAIT LOOP
1567  1374                      TAD     C1101   /AND ADJUST BLOCK TOTAL FOR
1570  3026                      DCA     BLOCKS  /1102(OCTAL) BLOCKS.
1571  5772                      JMP I   .+1
1572  1306              F10BAK, F10RTN
1573  7161              M617,   -617
1574  1101              C1101,  1101
```

```
                1600            *1600
                        /THE MARK TRACK HAS BEEN WRITTEN, AND TAPE IS
                        /MOVING FORWARD IN THE FORWARD END ZONE. STOP
                        /THE TAPE AND SEE IF THERE ARE ANY TAPES LEFT TO
                        /MARK--IF SO GO DO THEM, ELSE TELL OPERATOR TO THROW THE
                        /"NORMAL/WRTM/RDTM" SWITCH TO "NORMAL"
                        /HE WILL THEN CONTINUE AFTER THIS ACTION

                        /KILL WRITE, STOP TAPE

1600  1154      MWTM,   TAD   DT0070    /STOP TAPE WITH SELECT ERROR
1601  1030              TAD   DTA       /LOAD DTA INTO ORDER
1602  6766              DTCX            /EXECUTE THE ABOVE
1603  4303              JMS     NUDTA   /ANY MORE DTAS TO MARK?
1604  5702              JMP I   DOMARK  /YES; GO MARK THEM

                        /MESSAGE TO OPERATOR

1605  4531              JMS I TYPE      /NO: BACK TO FIRST DTA AND CONTINUE
1606  2305              2305  /SE
1607  2440              2440  /T
1610  2327              2327  /SW
1611  1124              1124  /IT
1612  0310              0310  /CH
1613  4024              4024  / T
1614  1740              1740  /O
1615  1617              1617  /NO
1616  2215              2215  /RM
1617  0114              0114  /AL
1620  0000              0000  /END
1621  4532              JMS I TYPIN     /WAIT FOR CR

                        /REVERSE TAPE FOR A FEW SECONDS TO GUARANTEE
                        /BLOCK MARK SECT WILL BE UNDER THE HEAD

1622  1164      PSER,   TAD   DT0600    /REVERSE, MOVE, GO
1623  1030              TAD   DTA       /ADD DTA TO ORDER
1624  6766              DTCX            /CLEAR TCU,GET MOVING IN REVERSE

                        /STALL A FEW SECONDS

1625  1104              TAD   M300      /AROUND 2 SECONDS
1626  3021              DCA   W2        /MAJOR STALL
1627  2020              ISZ   W1        /MINOR STALL
1630  5227              JMP   .-1       /LOOP MINOR
1631  2021              ISZ   W2        /MAJOR STALL
1632  5227              JMP   .-3       /LOOP MAJOR
```

```
                        /TAPE OUT ON MARK TRACK NOW, TURN AND GET IT
                        /MOVING FORWARD, AT THIS POINT, THE LAST REVERSE
                        /BLOCK NUMBER WILL BE WRITTEN UNTILL END ZONE IS
                        /REACHED, THEREFORE, WHEN THE BOUNCE OUT OF THE END
                        /ZONE TAKES PLACE, THE SYSTEM WILL BE ABLE TO SYNC ON
                        /THE REVERSE BLOCK NUMBER TO WRITE THE REST OF
                        /THE BLOCK NUMBERS AND KNOWN GOOD DATA IN REVERSE.
                        /THIS PROCESS WILL ELIMINATE A NEEDLESS REWIND AND
                        /KEEP THE ENTIRE PROCESS TO TWO COMPLETE PASSES

                        /WRITE LAST REVERSE BLOCK NUMBER  GOING FORWARD

1633  1161                      TAD   DT0210    /FORWARD, SEARCH, GO
1634  1030                      TAD   DTA       /ADD IN THE DTA
1635  6766                      DTCX            /CLEAR STATUS "A" AND RELOAD IT
1636  1036                      TAD   C1        /PHASE 1 ERROR
1637  3032                      DCA   PHASE     /FOR ERROR ROUTINE

                        /WAIT HERE FOR DECTAPE FLAG. CHECK ALSO FOR ERRORS
                        /SET BLOCK NUMBER (REVERSE) INTO FORM

1640  1026                      TAD   BLOCKS    /INTO AC WITH LAST BLOCK NUMBER
1641  4524                      JMS I MESS      /CONVERT BLOCK NUMBER FOR TAPE

                        /INTERRUPTED? ERROR?

1642  6772                      DTRB            /READ STATUS "B"
1643  7010                      RAR             /DECTAPE FLAG TO LINK
1644  7620                      SNL   CLA       /FLAG SET?
1645  5242                      JMP   .-3       /NO: CONTINUE WAIT

                        /BLOCK FOUND, SWITCH TO READ DATA WITH WC ONE LESS THAN
                        /NUMBER OF WORDS TO BE READ. READ TILL WC=0

1646  1156                      TAD   DT0130    /TO SET STATUS "A" INTO
1647  6764              RCYBR,  DTXA            /THE READ DATA MODE
1650  7240                      CLA   CMA       /SUBTRACT 1 FROM TOTAL
1651  1033                      TAD   TOTAL     /GIVING TOTAL-1 (HO HO)
1652  7040                      CMA             /INVERT FOR ISZ
1653  3534                      DCA I WC        /SET WC
1654  1041                      TAD   C4        /NOP
1655  3512                      DCA I CA        /JIMMIED TO DO NOTHING
1656  6772                      DTRB            /READ "B" REGISTER
1657  0062                      AND   C1000     /ISOLATE END ZONE BIT
1660  7640                      SZA   CLA       /END ZONE?
1661  5701                      JMP I GDBLK     /YES; GO AND WRITE THE BLOCK NUMBERS
1662  1534                      TAD I WC        /WAIT TILL WORD COUNT ZERO
1663  7640                      SZA   CLA       /EQUAL TO ZERO?
1664  5254                      JMP   .-10      /NO: LOOP AGAIN
```

```
                        /END OF BLOCK FOUND, WRITE JUNK AND REVERSE BLOCK NUMBER

1665  1102              TAD   M14       /12 WORDS TO BE WRITTEN
1666  3534              DCA I WC        /TO WORD COUNT REG.
1667  1117              TAD   FORMB     /FORM TO CA
1670  3512              DCA I CA        /OF NUMBERING FORM
1671  1154              TAD   DT0070    /SWITCH TO WRITE ALL
1672  6764              DTXA            /MODE.

                        /LOOK FOR THE DECTAPE FLAG INDICATING ANOTHER RECYCLE

1673  6772              DTRB            /NO: GET "B" AGAIN
1674  7010              RAR             /FLAG TO LINK
1675  7620              SNL   CLA       /FLAG SET?
1676  5273              JMP   .-3       /NO: BE PATIENT, HAST NOT.
1677  1154              TAD   DT0070    /TO SWITCH TO READ DATA
1700  5247              JMP   RCYBR
1701  2000  GDBLK,      DOBLK
1702  1400  DOMARK,     STMK    /POINTER TO START OF MARK ROUTINE

                        /SUBROUTINE TO GET NEXT DTA UNIT # FROM INPUT LIST OR
                        /RECYCLE TO FIRST UNIT IF ALL HAVE BEEN PROCESSED UP TO
                        /THIS POINT--CALL SEQUENCE
                        /       JMS NUDTA       /CALL THE ROUTINE
                        /       (RETN1)         /RETURNS HERE IF MORE DTAS TO PROCESS
                        /       (RETN2)         /RETURNS HERE IF END OF LIST
                        /END OF LIST MEANS RESET TO FIRST AND RETURN TO (RETN2)
                        /RETURN IS  WITH DTA SET TO NEW VALUE AND AC#0

1703  0000  NUDTA,      0
1704  1724              TAD I   LSTPT   /GET CURRENT VALUE OF DTA LIST PTR
1705  3323              DCA     TBUFPT  /STORE IT AS TEM. BUF. PTR.
1706  1723              TAD I   TBUFPT  /GET A DTA # FROM THE LIST
1707  0045              AND     C0007   /ISOLATE LOW ORDER DIGIT
1710  7640              SZA     CLA     /IS IT 7777?
1711  5316              JMP     LSTEND  /YES: END OF LIST
1712  1723              TAD I   TBUFPT  /NO: GET IT BACK
1713  3030              DCA     DTA     /AND STORE AS NEW DTA #
1714  2724              ISZ I   LSTPT   /INCREMENT LIST POINTER
1715  5703              JMP I   NUDTA   /RETURN
                        /COMES HERE AT END OF LIST TO RESET PTRS AND RETN TO CALL+2
1716  2303  LSTEND,     ISZ     NUDTA   /INCREMENT RETURN POINTER
1717  1722              TAD I   STRTPT  /GET ADR. OF START OF LIST
1720  3724              DCA I   LSTPT   /STORE TO RE-INITIALIZE LIST PTR.
1721  5304              JMP     NUDTA+1 /GO GET FIRST DTA # AND RETURN

1722  1176  STRTPT,     DBUFAD  /POINTER TO START OF DTA LIST
1723  0000  TBUFPT,     0       /TEM. STORAGE FOR BUF. PTR.
1724  1175  LSTPT,      DBUFPT  /POINTER TO CURRENT VALUE OF DTA LIST PTR
1725  0000  DTABUF,     0       /START OF DTA # LIST - MAX. 9 WORDS
```

```
                2000            *2000
                                /GO INTO SEARCH IN REVERSE MODE LOOKING FOR
                                /THE LAST BLOCK NUMBER, WHEN FOUND, SYNC THE SYSTEM
                                /AND WRITE ALL DATA AND BLOCK NUMBERS

2000  4527      DOBLK,  JMS I TURN      /INTO REVERSE AND SEARCH MODE
2001  1026              TAD   BLOCKS    /TO SET UP
2002  3027              DCA   BLOCKA    /FOR BLOCK DECREMENTING
2003  1037              TAD   C2        /PHASE 2 ERROR
2004  3032              DCA   PHASE     /FOR ERROR ROUTINE

                        /LOOK FOR INTERRUPT INDICATING BLOCK NUMBER

2005  4533              JMS I WAIT      /FOR DECTAPE FLAG

                        /SWITCH TO WRITE ALL, SYSTEM NOW IN SYNC

2006  1157              TAD   DT0140    /SWITCH TO WRITE ALL
2007  6764              DTXA            /EXECUTE ORDER
2010  1265      NEXTBN, TAD   ADF3      /ADDRESS OF FIRST 3 WORDS INCLUDING
2011  3512              DCA I CA        /THE FORWARD CHECKSUM TO BE WRITTEN
2012  1076              TAD   M3        /NUMBER OF WORDS TO BE WRITTEN
2013  3534              DCA I WC        /TO WORD COUNT
2014  4234              JMS   CEZN      /CHECK FOR END ZONE
2015  1534              TAD I WC        /CHECK FOR WC=0
2016  7640              SZA   CLA       /=0?
2017  5214              JMP   .-3       /NOPE: TRY AGAIN
2020  6764              DTXA            /YUP: CLEAR THE FLAG

                        /WRITE DATA TRACK, REMEMBER CORRECT DATA IS BEING WRITTEN

2021  1033              TAD   TOTAL     /ONE FROM TOTAL NUMBER
2022  7041              CIA             /OF WORDS FOR COUNTING
2023  3534              DCA I WC        /DATA WORDS WRITTEN
2024  1271              TAD   AD7777    /ADDRESS OF SEVENS
2025  3512              DCA I CA        /DATA TO BE WRITTEN

                        /MONITOR WORD COUNT FOR A ZERO READING
                        /SOME OF THIS TIME IS USED TO SET THE NEXT
                        /BLOCK NUMBER INTO THE FORM.

2026  1027              TAD   BLOCKA    /CURRENT BLOCK NUMBER
2027  4524              JMS I MESS      /CONVERT INTO FORM
2030  7240              CLA   CMA       /TO DECREMENT
2031  1027              TAD   BLOCKA    /THE BLOCK COUNT
2032  3027              DCA   BLOCKA    /DOWN TO ZERO
2033  5242              JMP   CEZB      /BYPASS FOLLOWING ROUTINE

                        /CHECK FOR END ZONE
2034  0000      CEZN,   0
2035  6772              DTRB            /READ STATUS "B"
2036  0062              AND   C1000     /ISOLATE END ZONE
2037  7650              SNA   CLA       /HAVE IT?
2040  5634              JMP I CEZN      /NOT EZ, RETURN
2041  5664              JMP I GOBN      /COMPLETED
```

```
                    /CHECK HERE ALSO TO SEE IF END ZONE, INDICATING
                    /THAT THE LAST BLOCK HAS BEEN WRITTEN

2042  4234  CEZB,   JMS   CEZN      /END ZONE?

                    /LOOK FOR WORD COUNT AS BEING EQUAL TO ZERO

2043  1534          TAD I WC        /WC TO C(AC)
2044  7650          SNA   CLA       /END OF DATA WRITE?
2045  5251          JMP   WBN       /YES; GO TO WRITE BLOCK NUMBER
2046  1271          TAD   AD7777    /RESET CURRENT ADDRESS COUNT
2047  3512          DCA I CA        /DON'T LET THE CA ADVANCE TO
2050  5242          JMP   CEZB      /MUCH

                    /DATA HAS BEEN WRITTEN, NOW WRITE REVERSE
                    /BLOCK NUMBER, FORWARD BLOCK NUMBER, AND REVERSE
                    /CHECKSUM. (12 WORDS)

2051  6764  WBN,    DTXA            /CLEAR OUT DECTAPE FLAG
2052  1102          TAD   M14       /WILL WRITE 12 WORDS
2053  3534          DCA I WC        /FOR THIS BIT
2054  1116          TAD   FORMA     /FROM A FORM CONTAINING
2055  3512          DCA I CA        /BLOCK NUMBERS

                    /WAIT FOR END

2056  4234          JMS   CEZN      /END ZONE?
2057  1534          TAD I WC        /NO: SEE IF DONE THE WRITE
2060  7640          SZA   CLA       /DONE YET ?
2061  5256          JMP   .-3       /NO: PATIENCE IS A VIRTUE????
2062  6764          DTXA            /RESET THE CURRENT FLAG
2063  5213          JMP   NEXTBN    /YES; GO RECYCLE COMPLETLY
2064  2200  GOBN,   DBN

                    / FIRST 3 WORDS TO BE WRITTEN

2065  2065  ADF3,   .
2066  0000          0000
2067  0000          0000
2070  0077          0077

                    /DATA TO BE WRITTEN ON TAPE (REVERSE)

2071  2071  AD7777, .
2072  7777          7777
2073  7777          7777
2074  7777          7777
2075  7777          7777
                    /CHECK IF ALL DTAS ARE DONE BEFORE RESTARTING

2076  4701  SETDTA, JMS I   GDTA    /ALL DTAS DONE?
2077  5702          JMP I   CONTNU  /NO: BACK TO WRITE BLOCK #S ON NEXT
2100  5515          JMP I   IT      /YES; GO ASK "DIRECT?"
2101  1703  GDTA,   NUDTA  /POINTER TO SUBR FOR GETTING NEXT UNIT #
2102  1622  CONTNU, PSER    /POINTER TO START OF BLOCK # WRITE ROUTINE
```

```
                2200            *2200
                                /VERIFY THE TAPE AS BEING WRITTEN CORRECTLY
                                /WITH DATA AND BLOCK NUMBERS. THE INFORMATION WRITTEN
                                /WAS WRITTEN IN SUCH A WAY AS TO BE CORRECT
                                /UPON READING IT BACK


                                /TURN TAPE AND HAVE IT GOING FORWARD

2200  1310      DBN,      TAD     ISZV        /RESET INCREMENT
2201  3250                DCA     VISZ        /BLOCK NUMBERS FORWARD
2202  3114                DCA     FCON        /WILL BE ZEROS FORWARD
2203  3020                DCA     W1          /FIRST BLOCK NUMBER FORWARD
2204  1047                TAD     C0400       /TURN TO GO FORWARD
2205  4527      DBNAUX,   JMS I   TURN
2206  1040                TAD     C3          /ERROR IN PHASE 3
2207  3032                DCA     PHASE       /FOR ERROR ROUTINE

                                /SET SOME OF THE CONTROL REGS

2210  3534      DAB,      DCA I   WC          /WORD COUNT DON'T CARE
2211  1166                TAD     ADBA        /SOME WHERE UP ABOVE
2212  3512                DCA I   CA          /TO GET BLOCK NUMBERS

                                /WAIT FOR INTERRUPT

2213  4533                JMS I   WAIT        /INTERRUPT
2214  1020                TAD     W1          /FIRST OR NEXT BLOCK NUMBER
2215  7041                CIA                 /TO COMPARE
2216  1566                TAD I   ADBA        /GET THE BLOCK NUMBER
2217  7640                SZA     CLA         /COMPARE OK?
2220  5312                JMP     BLKERZ      /BLOCK ERROR FOUND

                                /BLOCK COMPARES, NOW CHECK DATA

2221  1152                TAD     DT0030      /TO SWITCH INTO READ
2222  6764                DTXA                /DATA MODE
2223  3534                DCA I   WC          /DON'T CARE ABOUT THE WC
2224  1167      CTST,     TAD     ADWA        /FOR COMPARING
2225  3512                DCA I   CA          /FROM TAPE

                                /EVERY TIME THE WORD COUNT MOVES
                                /A DATA TRANSFERE HAS BEEN COMPLETED,
                                /MAKE SURE THAT THE INFORMATION IS OK

2226  1534                TAD I   WC          /GET WORD COUNT
2227  7650                SNA     CLA         /STILL AT ZERO?
2230  5237                JMP     CEFR        /YES; SEE IF AT END
2231  1114                TAD     FCON        /NO: SEE IF DATA
2232  7041                CIA                 /IS SAME AS WRITTEN
2233  1570                TAD I   ADWAB       /RECEIVED DATA
2234  7640                SZA     CLA         /SAME?
2235  5342                JMP DTAR            /DATA ERROR FOUND
2236  3534                DCA I   WC          /YES; RESET WORD COUNT
```

```
                        /CHECK FOR DECTAPE FLAG INDICATING END OF
                        /BLOCK OR ERROR

2237  6772  CEFR,  DTRB              /READ "B" REGISTER
2240  7510          SPA               /ERROR?
2241  5361          JMP   PARIR       /PARITY ERROR, I GUESS

                        /NO ERROR, END OF BLOCK?

2242  7010          RAR               /FLAG TO THE LINK
2243  7620          SNL   CLA         /END?
2244  5224          JMP   CTST        /NO: CONTINUE CHECKING
2245  1152          TAD   DT0030      /CLEAR DECTAPE FLAG
2246  6764          DTXA              /AND RETURN TO SEARCH

                        /END OF BLOCK, SEE IF END OF TAPE

2247  1020          TAD   W1          /BLOCK NUMBER JUST TESTED
2250  2020  VISZ,  ISZ   W1          /+1 OR -1 TO BLOCK COUNT
2251  7410          SKP
2252  7402          HLT               /ABSOLUTE PANIC
2253  7041          CIA               /TO BE COMPARED WITH
2254  1026          TAD   BLOCKS      /TOTAL BLOCKS
2255  7640          SZA   CLA         /LAST?
2256  5210          JMP   DAB         /NO, DO ANOTHER BLOCK


                        /HERE PUT IN THE REVERSE CHECK

2257  6771  DDSF,  DTSF              /WAIT FOR ANY FLAG TO APPEAR
2260  5257          JMP   .-1         /NOT YET
2261  7300          CLA   CLL         /RID  AC OF GARBAGE
2262  6772          DTRB              /READ THE "B" REGISTER
2263  0062          AND   C1000       /BETTER BE END ZONE
2264  7650          SNA   CLA         /IS IT?
2265  5363          JMP   LNE         /LAST INTERRUPT NOT END ZONE
2266  6766          DTCX              /YUP: A OK
```

```
                /BLOCK NUMBERS AND DATA HAVE  BEEN CHECKED FORWARD
                /AND ARE OK, USING THE ABOVE ROUTINE FOR CHECKING
                /RESET A FEW THINGS AND CHECK IN REVERSE

                /WAS COMPLETION FOUND FORWARD? IF SO GO CHECK
                /IN REVERSE; IF NOT GO SEE IF ALL TAPES HAVE BEEN CHECKED.

2267  1114              TAD   FCON      /IF 0'S, IT WAS FWD
2270  7640              SZA   CLA       /FWD?
2271  5711              JMP I   FINCHK  /NO: REVERSE-SEE IF ALL DTAS DONE

                /RESET THE ABOVE ROUTINE TO READ IN REVERSE

2272  7040              CMA             /DATA WILL BE AS WRITTEN
2273  3114              DCA   FCON      /I.E., 7777'S
2274  1301              TAD   SJMP      /INSTEAD OF INCREMENTING
2275  3250              DCA   VISZ      /WE WILL DECREMENT BLOCK NUMBERS
2276  1026              TAD   BLOCKS    /STARTING WITH THE HIGHEST
2277  3020              DCA   W1        /AND WILL WORK TO ZERO
2300  5205              JMP   DBNAUX    /ALL SET, TRAVEL ONWARD

                /RETURN HERE AFTER EACH BLOCK FOR CHECKING WHEN LAST BLOCK
                /HAS BEN PROCESSED???????????

2301  5302    SJMP,     JMP   .+1
2302  7450              SNA             /IF AC = 0, WE ARE DONE
2303  5257              JMP   DDSF      /AND NEXT FLAG SHOULD BE END ZONE
2304  7041              CIA             /OTHERWISE, SUBTRACT ONE FROM
2305  7041              CMA             /BLOCKS GIVING BLOCKS-1,,,,,?
2306  3020              DCA   W1        /NOT DONE
2307  5213              JMP   DAB       /GO DO ANOTHER BLOCK

2310  2020    ISZV,     ISZ   W1        /VARIABLE TAG
2311  2076    FINCHK,   SETDTA
```

```
                        /BLOCK ERROR FOUND

2312  1030    BLKERZ, TAD   DTA       /TO RESET TAPE
2313  6766            DTCX            /MOTION
2314  1566            TAD I ADBA      /GET BAD BLOCK NUMBER
2315  4533            JMS I TYOCT     /AND TYPE IT OUT
2316  4331            JMS   TYSB      /TYPE "SHOULD BE"
2317  1023            TAD   W1        /GOOD BLOCK NUMBER
2320  4533            JMS I TYOCT     /TYPE IT OUT
2321  4531            JMS I TYPE
2322  4002            4002  / B
2323  1413            1413  /LK
2324  4005            4005  / E
2325  2243            2243  /R CR
2326  4509            4500  /LF+END
2327  5730    DBERZ,  JMP I .+1
2330  0725            ZCOM

                        /COMMON ROUTINE

2331  0000    TYSB,   0
2332  4531            JMS I TYPE
2333  4023            4023  / S
2334  1017            1017  /HO
2335  2514            2514  /UL
2336  0440            0440  /D
2337  0205            0205  /BE
2340  4000            4000  / 0
2341  5731            JMP I TYSB

                        /DATA ERROR

2342  1030    DTAR,   TAD   DTA       /TO STOP TAPE
2343  6766            DTCX            /MOTION
2344  1567            TAD I ADWA      /GET THE BAD WORD
2345  4533            JMS I TYOCT
2346  4331            JMS TYSB        /TYPE "SHOULD BE"
2347  1114            TAD   FCON      /GOOD WORD
2350  4533            JMS I TYOCT     /TYPE IT OUT
2351  4531            JMS I TYPE
2352  4004            4004  /D
2353  0124            0124  /AT
2354  0140            0140  /A
2355  0522            0522  /ER
2356  4543            4543  /CR+LF
2357  0000            0000  /END
2360  5327            JMP   DBERZ
```

```
                        /PARITY ERROR FOUND

2361  5762    PARIR,    JMP I .+1
2362  0642              ERROR              /MAIN ERROR ROUTINE

                        /LAST INTERRUPT WAS NOT END ZONE

2363  4531    LNE,      JMS I TYPE
2364  1401              1401  /LA
2365  2324              2324  /ST
2366  4011              4011  / I
2367  1624              1624  /NT
2370  4016              4016  / N
2371  1724              1724  /OT
2372  4005              4005  / E
2373  1724              1724  /OT
2374  4345              4345  /LF+CR
2375  0000              0000  /END
2376  5327              JMP   DBERR
```

```
                2400            *2400
                                / TYPE OUT THE DTA UNIT NUMBER AND THE FIRST 12 BLOCK
                                /NUMBERS IN EITHER DIRECTION, IF RDR, IN REVERSE
                                /IF RDF, TYPE THEM OUT GOING IN THE FORWARD
                                /DIRECTION FROM THE BEGINING OF TAPE

        2400    1047    RDFA,   TAD     C0400   /DIRECTION FOR TURNING
        2401    3251            DCA     SAVEIT  /STORE DIRECTION FOR NEXT DTA UNIT
        2402    1251            TAD     SAVEIT  /GET DIRECTION FOR TURNING
        2403    4527            JMS I   TURN    /AROUND
        2404    1102            TAD     M14     /READ 12 BLOCK
        2405    3022            DCA     W3      /COUNTER
        2406    1110            TAD     BADD    /ADDRESS OF BUFFER
        2407    3011            DCA     X2      /TO AUTO INDEX 2
        2410    1107            TAD     ADW3    /ADDRESS OF W2
        2411    3512            DCA I   CA      /FOR DATA XFER
        2412    4533            JMS I   WAIT    /FOR BLOCK INTERRUPT
        2413    1021            TAD     W2      /BLOCK NUMBER
        2414    3411            DCA I   X2      /STORE BLOCK NUMBER
        2415    2022            ISZ     W3      /TOTAL = 12?
        2416    5212            JMP     .-4     /NO: GRAB NEXT
        2417    1030            TAD     DTA     /KILL TAPE MOTION
        2420    6766            DTCX            /HERE

                                /TYPE OUT BLOCK NUMBERS AND DTA UNIT #

        2421    4531            JMS I   TYPE    /TYPE "DTA"
        2422    0424            0424    /DT
        2423    0140            0140    /A
        2424    0000            0000    /END
        2425    1030            TAD     DTA     /GET UNIT #
        2426    4530            JMS I   TYOCT   /AND TYPE IT OUT
        2427    4531            JMS I   TYPE
        2430    4345            4345    /CR&LF
        2431    0000            0000    /END
        2432    1102            TAD     M14     /WILL TYPE ALL
        2433    3020            DCA     W1      /TWELVE WORDS
        2434    1110            TAD     BADD    /ADDRESS OF BLOCK
        2435    3011            DCA     X2      /NUMBERS TO INDEX 2
        2436    1411            TAD I   X2      /FIRST OR NEXT BLOCK
        2437    4530            JMS I   TYOCT   /TYPE IT OUT
        2440    4531            JMS I   TYPE    /CR AND LINE FEED
        2441    4345            4345    /CR+LF
        2442    0000            0000
        2443    2020            ISZ     W1      /COMPLETE?
        2444    5236            JMP     .-6     /NO
        2445    4652            JMS I   NEWDTA  /YES: ANY MORE DTAS?
        2446    5202            JMP     RDFA+2  /YES: GO GET BLOCK #S
        2447    5515            JMP I   IT      /NO: GO ASK FOR "DIRECT?"
        2450    5201    RDR,    JMP     RDFA+1  /OTHER DIRECTION

        2451    0000    SAVEIT, 0               /TEM, STORAGE FOR DIRECTION
        2452    1703    NEWDTA, NUDTA           /POINTER TO SUBR, TO GET A NEW DTA UNIT #
```

```
                        /INPUT BUFFER FOR THE TELETYPE.
                        /NOTE ,,,,,,,THIS MUST BE AT THE END OF THE PROGRAM

2453  0000              BUFFER, 0000

                        $
```

```
V2        0534
V3        0540
V4        0541
VALCHK    1023
VAR1      0034
VAR2      0035
VISZ      2250
W1        0020
W2        0021
W3        0022
W4        0023
W5        0024
W6        0025
WAIT      0133
WBN       2051
WC        0134
WDZ       1435
WEZF      1501
WLMRF     1526
X1        0010
X2        0011
ZCOM      0725
ZEOT      0661
ZFORM     0757
ZMKTK     0674
ZPAR      0704
ZSEL      0712
ZTIM      0720
```

ERRORS DETECTED: 0

LINKS GENERATED: 0

RUN-TIME: 16 SECONDS

2K CORE USED

```
AD7777  2071        CERR    0330        DZ      1535        NEXTBN  2010
ADBA    0166        CEZ     1407        ERROR   0642        NTYRTN  0263
ADF3    2065        CEZB    2042        ERX     0031        NUDTA   1703
ADW2    0106        CEZN    2034        EZM     1554        OKCR    1044
ADW3    0107        CIZ     1371        F10BAK  1572        PARIR   2361
ADWA    0167        COMCHK  1156        F10PAT  1560        PFORM   0773
ADWAB   0170        COMCTR  1173        F10RTN  1306        PHASE   0032
BADD    0110        COMPAR  0113        FBM     1522        PSER    1622
BFR     0111        COMPRE  0310        FCON    0114        Q1      0400
BINCON  0171        CONC    0543        FEZ     1541        Q2      0410
BLKERZ  2312        CONTNU  2102        FINCHK  2311        Q3      0422
BLOCKA  0027        CRCHK   1015        FORM    0527        Q4      0434
BLOCKS  0026        CRCOD   0072        FORM10  1375        QU      0143
BUFFER  2453        CRFLAG  1146        FORMA   0116        QU1     0120
C0007   0045        CTST    2224        FORMB   0117        QU2     0121
C0017   0042        DAB     2210        GDBLK   1701        QU3     0122
C0030   0046        DBERZ   2327        GDBN    2064        QU4     0123
C0070   0043        DBN     2200        GDTA    2101        QUX     0447
C0077   0044        DBNAUX  2205        GETDTA  1145        RCYBR   1647
C0430   0047        DBUFAD  1176        GRZ     1550        RDFA    2400
C0700   0050        DBUFPT  1175        IBS     1242        RDR     2450
C0777   0632        DCTR    1174        IBZ     1516        REPEAT  1147
C1      0036        DDSF    2257        INBLSY  1424        RERR    0610
C1000   0262        DIV3    1244        INIT    1061        REZ     1512
C1101   1574        DNC     1203        INIT1   1055        RSEND   0135
C1620   0263        DNUM    1177        ISZV    2310        RSTSM   1353
C2      0037        DOBLK   2000        IT      0115        RSYC    0566
C201    0052        DOCOMP  1164        LETK    0073        SAVEIT  2451
C203    0051        DOMARK  1702        LFCOD   0074        SETDTA  2076
C213    0053        DT0030  0152        LNE     2363        SJMP    2301
C217    0253        DT0060  0153        LSTEND  1716        SPCOD   0105
C215    0254        DT0070  0154        LSTPT   1724        STALL   0600
C245    0255        DT0100  0155        M14     0102        START   1000
C262    0354        DT0130  0156        M144    0103        STMK    1400
C261    0355        DT0140  0157        M2      0075        STRTPT  1722
C267    0256        DT0200  0160        M3      0076        STX     0126
C270    0257        DT0210  0161        M300    0104        TBUFPT  1723
C271    0260        DT0360  0162        M4      0077        TFORM   0737
C277    0261        DT0510  0163        M40     0251        TOTAL   0033
C3      0040        DT0600  0164        M6      0100        TRN     0613
C343    0252        DT0610  0165        M617    1573        TURN    0127
C4      0041        DTA     0030        M7      0101        TW1     0374
C6060   0375        DTABUF  1725        MARK    1200        TYCT    0334
C7002   0264        DTAR    2342        MES     0454        TYCT1   0367
C7720   0265        DTCA    6762        MESAGE  0200        TYOCT   0130
C7714   0266        DTCX    6766        MESS    0124        TYPE    0131
C7761   0267        DTLB    6774        MINI    0633        TYPECH  0220
C7772   0267        DTRA    6761        MSRGHT  0217        TYPIN   0132
C7775   0271        DTRB    6772        MTP     0244        TYPN    0256
CA      0112        DTRK    1447        MWAIT   0125        TYQU    1003
CDTRD   1357        DTSF    6771        MWTM    1600        TYSB    2331
CEFR    2237        DTXA    6764        NEWDTA  2452        V1      0533
```

Title		Tech Tip Number
PROBLEM WITH TC01-TU55 DECTAPE FORMATTER		TC01-TT-4
Processor Applicability: All 8's	Author K.Wunderlich Rev Ø Approval B.Hansen Date 2/6/74	Cross Reference: TU55-TT-7

The program TC01-TU55 Dectape Formatter, DEC-08-EUFB will write the correct mark track and timing track on the dectapes first specified by the user. However, following the statement "set switch to normal" and after the user resets the switch and types carriage return, the first dectape specified will go in reverse for some time and then switch to the forward direction writing the last reverse block number. At this point in time due to the current address not getting reset by the program after writing the Forward End Zone during the write timing and mark track pass location 1557 gets modified from ØØ4Ø to 044Ø. Location 1557 contains parts of the code for writing the Forward End Zone, therefore, any new dectapes puts on the drives after this will not have the correct forward end zone written on the mark track.

The error can be detected by the TD8-E Diagnostic, (Maindec-08-DHTDA) routine for checking the mark track and the routine to search and find all block numbers.

To avoid this the following change can be made:

Location	Old	New
1633	1161	4360
1760		0000
1761		1166
1762		3512
1763		1161
1764		5760

DEC-08-EUFB has been corrected and resubmitted to the Program Library as DEC-08-UDTFA-A.